
Abstract

The groupoids package for GAP4 provides functions for the computation with groupoids (categories with

every arrow invertible) and their morphisms; for graphs of groups, and graphs of groupoids. The most basic

structure introduced is that of magma with objects, followed by semigroup with objects, then monoid with

objects and �nally groupoid which is a group with objects.

It provides normal forms for Free Products with Amalgamation and for HNN-extensions when the initial

groups have rewrite systems and the subgroups have �nite index.

The groupoids package was originally implemented in 2000 (as GraphGpd) when the �rst author was

studying for a Ph.D. in Bangor.

The package was then renamed Gpd and version 1.07 was released in July 2011, to be tested with GAP

4.5. Gpd became an accepted GAP package in May 2015.

In April 2017 the package was renamed again, as groupoids.

Recent versions implement many of the constructions described in the paper [AW10] for automorphisms

of groupoids.

The latest version is 1.54 of 29th November 2017.

Bug reports, comments, suggestions for additional features, and offers to implement some of these, will all

be very welcome.

Please submit any issues at http://github.com/gap-packages/groupoids/issues/ or send an

email to the second author at c.d.wensley@bangor.ac.uk.

Chapter 1

Introduction

Groupoids are mathematical categories in which every arrow is invertible. The groupoids package

provides functions for the computation with groupoids and their morphisms; for graphs of groups and

graphs of groupoids. The package is far from complete, and development continues.

It was used by Emma Moore in her thesis [Moo01] to calculate normal forms for free products

with amalgamation, and for HNN-extensions when the initial groups have rewriting systems.

The latest version is 1.54 for GAP 4.8, released on 29th November 2017, and may be more recent

than the release on the GAP archive.

The package may be obtained as a compressed tar �le groupoids-1.54.tar.gz by ftp from one

of the following sites:

� any GAP archive, e.g. http://www.gap-system.org/Packages/groupoids.html;

� the package GitHub repository: https://gap-packages.github.io/groupoids.

The information parameter InfoGroupoids takes default value 1 which, for the bene�t of new

users, causes more messages to be printed out when operations fail. When raised to a higher value,

additional information is printed out.

Help is available in the usual way.

Example

gap> LoadPackage( "groupoids" );

For version 1.05 the package was completely restructured, starting withmagmas with objects and their

mappings, and building up to groupoids via semigroups with objects and monoids with objects. From

version 1.07 the package includes some functions to implement constructions contained in [AW10].

More functions will be released as soon as possible.

Once the package is loaded, it is possible to check the correct installation by running the test

suite of the package with the command ReadPackage("groupoids","tst/testing.g"); . Ad-

ditional tests may be run using ReadPackage("groupoids","tst/testextra.g");. (The �le

"tst/testall.g" is used for automated testing.)

You may reference this package by mentioning [BMPW02], [Moo01] and [AW10].

Additional information can be found on the Computational Higher Dimensional Algebra website

at: http://pages.bangor.ac.uk/~mas023/chda/intro.html.

Chapter 2

Many-object structures

The aim of this package is to provide operations for �nite groupoids. A groupoid is constructed from

a group and a set of objects. In order to provide a sequence of categories, with increasing structure,

mimicing those for groups, we introduce in this chapter the notions of magma with objects; semigroup

with objects and monoid with objects. The next chapter introduces morphisms of these structures. At

a �rst reading of this manual, the user is advised to skip quickly through these �rst two chapters, and

then move on to groupoids in Chapter 3.

For the de�nitions of the standard properties of groupoids we refer to P. Higgins' book �Categories

and Groupoids� [Hig05] (originally published in 1971, reprinted by TAC in 2005), and to R. Brown's

book �Topology� [Bro88], recently revised and reissued as �Topology and Groupoids� [Bro06].

2.1 Magmas with objects; arrows

A magma with objects M consists of a set of objects Ob(M), and a set of arrows Arr(M) together with
tail and head maps t;h : Arr(M)! Ob(M), and a partial multiplication � : Arr(M)! Arr(M), with
a�b de�ned precisely when the head of a coincides with the tail of b. We write an arrow a with tail u

and head v as (a : u! v).
When this multiplication is associative we obtain a semigroup with objects.

A loop is an arrow whose tail and head are the same object. An identity arrow at object u is a loop

(1u : u! u) such that a�1u = a and 1u �b= b whenever u is the head of a and the tail of b. WhenM

is a semigroup with objects and every object has an identity arrow, we obtain a monoid with objects,

which is just the usual notion of mathematical category.

An arrow (a : u! v) in a monoid with objects has inverse (a�1 : v! u) provided a � a�1 = 1u
and a�1 �a= 1v. A monoid with objects in which every arrow has an inverse is a group with objects,

usually called a groupoid.

2.1.1 MagmaWithObjects

. MagmaWithObjects(args) (function)

. SinglePieceMagmaWithObjects(magma, obs) (operation)

. ObjectList(mwo) (attribute)

. RootObject(mwo) (attribute)

6

The simplest construction for a magma with objectsM is to take a magma m and an ordered set s,

and form arrows (u;a;v) for every a inm and u;v in s. Multiplication is de�ned by (u;a;v)�(v;b;w) =
(u;a � b;w). In this package we prefer to write (u;a;v) as (a : u! v), so that the multiplication rule

becomes (a : u! v)� (b : v! w) = (a�b : u! w).
Any �nite, ordered set is in principle acceptable as the object list of M, but most of the time we

�nd it convenient to restrict ourselves to sets of non-positive integers.

This is the only construction implemented here for magmas, semigroups, and monoids with ob-

jects, and these all have the property IsDirectProductWithCompleteDigraph. There are other

constructions implemented for groupoids.

The output from function MagmaWithObjects lies in the categories IsDomainWithObjects,

IsMagmaWithObjects and CategoryCollections(IsMultiplicativeElementWithObjects).

As composition is only partial, the output does not lie in the category IsMagma.

The root object of M is the �rst element in s.

Example

gap> tm := [[1,2,4,3],[1,2,4,3],[3,4,2,1],[3,4,2,1]];;

gap> Display( tm );

[ [ 1, 2, 4, 3 ],

[ 1, 2, 4, 3 ],

[ 3, 4, 2, 1 ],

[ 3, 4, 2, 1 ] ]

gap> m := MagmaByMultiplicationTable( tm );; SetName( m, "m" );

gap> m1 := MagmaElement(m,1);; m2 := MagmaElement(m,2);;

gap> m3 := MagmaElement(m,3);; m4 := MagmaElement(m,4);;

gap> M78 := MagmaWithObjects( m, [-8,-7] );

magma with objects :-

magma = m

objects = [ -8, -7 ]

gap> SetName( M78, "M78" );

gap> [ IsAssociative(M78), IsCommutative(M78) ];

[ false, false ]

gap> [ IsDomainWithObjects(M78), IsMagmaWithObjects(M78), IsMagma(M78) ];

[ true, true, false ]

gap> [ RootObject( M78 ), ObjectList( M78 ) ];

[ -8, [ -8, -7 ] ]

2.1.2 Arrow

. Arrow(mwo, elt, tail, head) (operation)

. ElementOfArrow(arr) (operation)

. TailOfArrow(arr) (operation)

. HeadOfArrow(arr) (operation)

Arrows in a magma with objects lie in the category IsMultiplicativeElementWithObjects.

An attempt to multiply two arrows which do not compose resuts in fail being returned. Each

arrow arr = (a : u ! v) has three components. The magma element a 2 m may be ac-

cessed by ElementOfArrow(arr). Similarly, the tail object u and the head object v may

be obtained using TailOfArrow(arr) and HeadOfArrow(arr) respectively. The operation

MultiplicativeElementWithObjects is a synonym for Arrow since this was used in older ver-

sions of the package.

Example

gap> a78 := Arrow( M78, m2, -7, -8 );

[m2 : -7 -> -8]

gap> a78 in M78;

true

gap> b87 := Arrow( M78, m4, -8, -7 );;

gap> [ ElementOfArrow( b87 ), TailOfArrow( b87 ), HeadOfArrow( b87 ) ];

[ m4, -8, -7 ]

gap> ba := b87*a78;; ab := a78*b87;; [ ba, ab ];

[ [m4 : -8 -> -8], [m3 : -7 -> -7] ]

gap> [ a78^2, ba^2, ba^3 ];

[ fail, [m1 : -8 -> -8], [m3 : -8 -> -8] ]

gap> ## this demonstrates non-associativity:

gap> [ a78*ba, ab*a78, a78*ba=ab*a78 ];

[ [m3 : -7 -> -8], [m4 : -7 -> -8], false ]

2.1.3 IsSinglePiece

. IsSinglePiece(mwo) (property)

. IsDirectProductWithCompleteDigraph(mwo) (property)

. IsDiscrete(mwo) (property)

If the partial composition is forgotten, then what remains is a digraph (usually with multiple edges

and loops). Thus the notion of connected component may be inherited by magmas with objects from

digraphs. Unfortunately the terms Component and Constituent are already in considerable use

elsewhere in GAP, so (and this may change if a more suitable word is suggested) we use the term

IsSinglePiece to describe a connected magma with objects. When each connected component has

a single object, and there is more than one component, the magma with objects is discrete.

Example

gap> IsSinglePiece( M78 );

true

gap> IsDirectProductWithCompleteDigraph( M78 );

true

gap> IsDiscrete( M78 );

false

2.2 Semigroups with objects

2.2.1 SemigroupWithObjects

. SemigroupWithObjects(args) (function)

. SinglePieceSemigroupWithObjects(sgp, obs) (operation)

. DomainWithSingleObject(dom, obj) (operation)

The constructions in section 2.1 give a SinglePieceSemigroupWithObjects when the magma

is a semigroup. In the example we use a transformation semigroup and 3 objects.

Example

gap> t := Transformation( [1,1,2,3] );;

gap> s := Transformation( [2,2,3,3] );;

gap> r := Transformation( [2,3,4,4] );;

gap> sgp := Semigroup( t, s, r );;

gap> SetName( sgp, "sgp<t,s,r>" );

gap> S123 := SemigroupWithObjects( sgp, [-3,-2,-1] );

semigroup with objects :-

magma = sgp<t,s,r>

objects = [ -3, -2, -1 ]

gap> [ IsAssociative(S123), IsCommutative(S123) ];

[ true, false ]

gap> t12 := Arrow( S123, t, -1, -2 );

[Transformation( [ 1, 1, 2, 3 ] ) : -1 -> -2]

gap> s23 := Arrow( S123, s, -2, -3 );

[Transformation( [ 2, 2, 3, 3 ] ) : -2 -> -3]

gap> r31 := Arrow( S123, r, -3, -1 );

[Transformation( [ 2, 3, 4, 4 ] ) : -3 -> -1]

gap> ts13 := t12 * s23;

[Transformation( [ 2, 2, 2, 3 ] ) : -1 -> -3]

gap> sr21 := s23 * r31;

[Transformation( [ 3, 3, 4, 4 ] ) : -2 -> -1]

gap> rt32 := r31 * t12;

[Transformation( [ 1, 2, 3, 3 ] ) : -3 -> -2]

gap> tsr1 := ts13 * r31;

[Transformation( [ 3, 3, 3 ] ) : -1 -> -1]

A magma, semigroup, monoid, or group can be made into a magma with objects by the addition of a

single object. The two are algebraically isomorphic, and there is one arrow (a loop) for each element

in dom. In the example we take the semigroup sgp of size 17 at the object 0.

Example

gap> S0 := DomainWithSingleObject( sgp, 0 );

semigroup with objects :-

magma = sgp<t,s,r>

objects = [ 0 ]

gap> t0 := Arrow( S0, t, 0, 0 );

[Transformation( [ 1, 1, 2, 3 ] ) : 0 -> 0]

gap> Size( S0 );

17

2.3 Monoids with objects

2.3.1 MonoidWithObjects

. MonoidWithObjects(args) (function)

. SinglePieceMonoidWithObjects(mon, obs) (operation)

The constructions in section 2.1 give a SinglePieceMonoidWithObjects when the magma is a

monoid. The example uses a �nitely presented monoid with 2 generators and 2 objects.

Example

gap> fm := FreeMonoid( 2, "f" );;

gap> em := One( fm );;

gap> gm := GeneratorsOfMonoid( fm );;

gap> mon := fm/[ [gm[1]^3,em], [gm[1]*gm[2],gm[2]] ];;

gap> M49 := MonoidWithObjects( mon, [-9,-4] );

monoid with objects :-

magma = Monoid( [ f1, f2 ] )

objects = [ -9, -4 ]

gap> ktpo := KnownTruePropertiesOfObject( M49 );

[ "IsDuplicateFree", "IsAssociative", "IsSinglePieceDomain",

"IsDirectProductWithCompleteDigraphDomain" ]

gap> catobj := CategoriesOfObject( M49 );;

[ "IsListOrCollection", "IsCollection", "IsExtLElement",

"CategoryCollections(IsExtLElement)", "IsExtRElement",

"CategoryCollections(IsExtRElement)",

"CategoryCollections(IsMultiplicativeElement)", "IsGeneralizedDomain",

"IsDomainWithObjects",

"CategoryCollections(IsMultiplicativeElementWithObjects)",

"CategoryCollections(IsMultiplicativeElementWithObjectsAndOnes)",

"CategoryCollections(IsMultiplicativeElementWithObjectsAndInverses)",

"IsMagmaWithObjects", "IsSemigroupWithObjects", "IsMonoidWithObjects" ]

2.4 Generators of magmas with objects

2.4.1 GeneratorsOfMagmaWithObjects

. GeneratorsOfMagmaWithObjects(mwo) (operation)

. GeneratorsOfSemigroupWithObjects(swo) (operation)

. GeneratorsOfMonoidWithObjects(mwo) (operation)

For a magma or semigroup with objects, the generating set consists of arrows (g : u! v) for every
pair of objects u;v and every generating element for the magma or semigroup.

For a monoid with objects, the generating set consists of two parts. Firstly, there is a loop at the

root object r for each generator of the monoid. Secondly, for each object u distinct from r, there are

arrows (1 : r! u) and (1 : u! r). (Perhaps only one of each pair is required?) Then

(e : u! v) = (1 : u! r)� (e : r! r)� (1 : r! v):

Example

gap> GeneratorsOfMagmaWithObjects( M78 );

[ [m1 : -8 -> -8], [m2 : -8 -> -8], [m3 : -8 -> -8], [m4 : -8 -> -8],

[m1 : -8 -> -7], [m2 : -8 -> -7], [m3 : -8 -> -7], [m4 : -8 -> -7],

[m1 : -7 -> -8], [m2 : -7 -> -8], [m3 : -7 -> -8], [m4 : -7 -> -8],

[m1 : -7 -> -7], [m2 : -7 -> -7], [m3 : -7 -> -7], [m4 : -7 -> -7] ]

gap> genS := GeneratorsOfSemigroupWithObjects( S123 );;

gap> Length( genS );

27

gap> genM := GeneratorsOfMonoidWithObjects( M49 );

[ [f1 : -9 -> -9], [f2 : -9 -> -9], [<identity ...> : -9 -> -4],

[<identity ...> : -4 -> -9] ]

gap> g1:=genM[2];; g2:=genM[3];; g3:=genM[4];; g4:=genM[5];;

gap> [g4,g2,g1,g3];

[ [<identity ...> : -4 -> -9], [f2 : -9 -> -9], [f1 : -9 -> -9],

[<identity ...> : -9 -> -4] ]

gap> g4*g2*g1*g3;

[f2*f1 : -4 -> -4]

2.5 Structures with more than one piece

2.5.1 UnionOfPieces

. UnionOfPieces(pieces) (operation)

. Pieces(mwo) (attribute)

A magma with objects whose underlying digraph has two or more connected components can be

constructed by taking the union of two or more connected structures. These, in turn, can be combined

together. The only requirement is that all the object lists should be disjoint. The pieces are ordered by

the order of their root objects.

Example

gap> N1 := UnionOfPieces( [ M78, S123 ] );

magma with objects having 2 pieces :-

1: M78

2: semigroup with objects :-

magma = sgp<t,s,r>

objects = [ -3, -2, -1 ]

gap> ObjectList( N1 );

[ -8, -7, -3, -2, -1 ]

gap> Pieces(N1);

[ M78, semigroup with objects :-

magma = sgp<t,s,r>

objects = [ -3, -2, -1 ]

]

gap> N2 := UnionOfPieces( [ M49, S0 ] );

semigroup with objects having 2 pieces :-

1: monoid with objects :-

magma = Monoid( [ f1, f2 ] )

objects = [ -9, -4 ]

2: semigroup with objects :-

magma = sgp<t,s,r>

objects = [ 0 ]

gap> ObjectList( N2 );

[ -9, -4, 0 ]

gap> N3 := UnionOfPieces( [ N1, N2] );

magma with objects having 4 pieces :-

1: monoid with objects :-

magma = Monoid( [ f1, f2 ] )

objects = [ -9, -4 ]

2: M78

3: semigroup with objects :-

magma = sgp<t,s,r>

objects = [ -3, -2, -1 ]

4: semigroup with objects :-

magma = sgp<t,s,r>

objects = [ 0 ]

gap> ObjectList( N3 );

[ -9, -8, -7, -4, -3, -2, -1, 0 ]

gap> Length( GeneratorsOfMagmaWithObjects( N3 ) );

50

gap> ## the next command returns fail since the object sets are not disjoint:

gap> N4 := UnionOfPieces( [ S123, DomainWithSingleObject( sgp, -2 ) ] );

fail



Chapter 3

Mappings of many-object structures

A homomorphism f from a magma with objects M to a magma with objects N consists of

� a map fO from the objects of M to those of N,

� a map fA from the arrows of M to those of N.

The map fA is required to be compatible with the tail and head maps and to preserve multiplication:

fA(a : u! v)� fA(b : v! w) = fA(a�b : u! w)

with tail fO(u) and head fO(w).
When the underlying magma of M is a monoid or group, the map fA is required to preserve

identities and inverses.

3.1 Homomorphisms of magmas with objects

3.1.1 MagmaWithObjectsHomomorphism

. MagmaWithObjectsHomomorphism(args) (function)

. HomomorphismFromSinglePiece(src, rng, hom, imobs) (operation)

. HomomorphismToSinglePiece(src, rng, images) (operation)

. MappingToSinglePieceData(mwohom) (attribute)

. PiecesOfMapping(mwohom) (attribute)

. IsomorphismNewObjects(src, objlist) (operation)

There are a variety of homomorphism constructors.

The simplest construction gives a homomorphism M ! N with both M and N con-

nected. It is implemented as IsMappingToSinglePieceRep with attributes Source, Range and

MappingToSinglePieceData. The operation requires the following information:

� a magma homomorphism hom from the underlying magma ofM to the underlying magma of N,

� a list imobs of the images of the objects ofM.

In the �rst example we construct endomappings of m and M78.

13

Example

gap> tup1 := [Tuple([m1,m2]), Tuple([m2,m1]), Tuple([m3,m4]), Tuple([m4,m3])];

gap> f1 := GeneralMappingByElements( m, m, tup1 );

gap> IsMagmaHomomorphism( f1 );

true

gap> hom1 := MagmaWithObjectsHomomorphism( M78, M78, f1, [-8,-7] );;

gap> Display( hom1 );

homomorphism to single piece magma: M78 -> M78

magma hom: <mapping: m -> m >

object map: [ -8, -7 ] -> [ -8, -7 ]

gap> [ Source( hom1 ), Range( hom1 ) ];

[ M78, M78 ]

gap> b87;

[m4 : -8 -> -7]

gap> im1 := ImageElm( hom1, b87 );

[m3 : -8 -> -7]

gap> i56 := IsomorphismNewObjects( M78, [-5,-6] );

magma with objects homomorphism :

[ [ IdentityMapping( m ), [ -5, -6 ] ] ]

gap> ib87 := ImageElm( i56, b87 );

[m4 : -5 -> -6]

gap> M65 := Range( i56);;

gap> SetName( M65, "M65" );

gap> j56 := InverseGeneralMapping( i56 );;

gap> ImagesOfObjects( j56 );

[ -7, -8 ]

gap> comp := j56 * hom1;

magma with objects homomorphism : M65 -> M78

[ [ <mapping: m -> m >, [ -7, -8 ] ] ]

gap> ImageElm( comp, ib87 );

[m3 : -8 -> -7]

A homomorphism to a connected magma with objects may have a source with several pieces, and so

is a union of homomorphisms from single pieces.

Example

gap> M4 := UnionOfPieces( [ M78, M65 ] );;

gap> images := [ MappingToSinglePieceData( hom1 )[1],

> MappingToSinglePieceData( j56 )[1] ];

[ [ <mapping: m -> m >, [ -8, -7 ] ], [ IdentityMapping( m ), [ -7, -8 ] ] ]

gap> map4 := HomomorphismToSinglePiece( M4, M78, images );

magma with objects homomorphism :

[ [ <mapping: m -> m >, [ -8, -7 ] ], [ IdentityMapping( m ), [ -7, -8 ] ] ]

gap> ImageElm( map4, b87 );

[m3 : -8 -> -7]

gap> ImageElm( map4, ib87 );

[m4 : -8 -> -7]

3.2 Homomorphisms of semigroups and monoids with objects

The next example exhibits a homomorphism between transformation semigroups with objects.

Example

gap> t2 := Transformation( [2,2,4,1] );;

gap> s2 := Transformation( [1,1,4,4] );;

gap> r2 := Transformation( [4,1,3,3] );;

gap> sgp2 := Semigroup( [ t2, s2, r2 ] );;

gap> SetName( sgp2, "sgp<t2,s2,r2>" );

gap> ## apparently no method for transformation semigroups available for:

gap> ## nat := NaturalHomomorphismByGenerators( sgp, sgp2 ); so we use:

gap> ## in the function flip below t is a transformation on [1..n]

gap> flip := function( t )

> local i, j, k, L, L2, n;

> n := DegreeOfTransformation( t );

> L := ImageListOfTransformation( t );

> if IsOddInt(n) then n:=n+1; L1:=Concatenation(L,[n]);

> else L1:=L; fi;

> L2 := ShallowCopy( L1 );

> for i in [1..n] do

> if IsOddInt(i) then j:=i+1; else j:=i-1; fi;

> k := L1[j];

> if IsOddInt(k) then L2[i]:=k+1; else L2[i]:=k-1; fi;

> od;

> return( Transformation( L2 ) );

> end;;

gap> smap := MappingByFunction( sgp, sgp2, flip );;

gap> ok := RespectsMultiplication( smap );

true

gap> [ t, ImageElm( smap, t ) ];

[ Transformation( [ 1, 1, 2, 3 ] ), Transformation( [ 2, 2, 4, 1 ] ) ]

gap> [ s, ImageElm( smap, s ) ];

[ Transformation( [ 2, 2, 3, 3 ] ), Transformation( [ 1, 1, 4, 4 ] ) ]

gap> [ r, ImageElm( smap, r ) ];

[ Transformation( [ 2, 3, 4, 4 ] ), Transformation( [ 4, 1, 3, 3 ] ) ]

gap> SetName( smap, "smap" );

gap> T123 := SemigroupWithObjects( sgp2, [-13,-12,-11] );;

gap> shom := MagmaWithObjectsHomomorphism( S123, T123, smap, [-11,-12,-13] );;

gap> it12 := ImageElm( shom, t12 );; [ t12, it12 ];

[ [Transformation( [ 1, 1, 2, 3 ] ) : -1 -> -2],

[Transformation( [ 2, 2, 4, 1 ] ) : -13 -> -12] ]

gap> is23 := ImageElm( shom, s23 );; [ s23, is23 ];

[ [Transformation( [ 2, 2, 3, 3 ] ) : -2 -> -3],

[Transformation( [ 1, 1, 4, 4 ] ) : -12 -> -11] ]

gap> ir31 := ImageElm( shom, r31 );; [ r31, ir31 ];

[ [Transformation( [ 2, 3, 4, 4 ] ) : -3 -> -1],

[Transformation( [ 4, 1, 3, 3 ] ) : -11 -> -13] ]

3.3 Homomorphisms to more than one piece

3.3.1 HomomorphismByUnion

. HomomorphismByUnion(src, rng, homs) (operation)

When f :M! N and N has more than one connected component, then f is a union of homomor-

phisms, one for each piece in the range.

Example

gap> N4 := UnionOfPieces( [ M78, T123 ] );

magma with objects having 2 pieces :-

1: semigroup with objects :-

magma = sgp<t2,s2,r2>

objects = [ -13, -12, -11 ]

2: M78

gap> h14 := HomomorphismByUnionNC( N1, N4, [ hom1, shom ] );

magma with objects homomorphism :

[ magma with objects homomorphism : M78 -> M78

[ [ <mapping: m -> m >, [ -8, -7 ] ] ], magma with objects homomorphism :

[ [ smap, [ -11, -12, -13 ] ] ] ]

gap> ImageElm( h14, a78 );

[m1 : -7 -> -8]

gap> ImageElm( h14, r31 );

[Transformation( [ 4, 1, 3, 3 ] ) : -11 -> -13]

3.3.2 IsInjectiveOnObjects

. IsInjectiveOnObjects(mwohom) (property)

. IsSurjectiveOnObjects(mwohom) (property)

. IsBijectiveOnObjects(mwohom) (property)

. IsEndomorphismWithObjects(mwohom) (property)

. IsAutomorphismWithObjects(mwohom) (property)

The meaning of these �ve properties is obvious.

Example

gap> IsInjectiveOnObjects( h14 );

true

gap> IsSurjectiveOnObjects( h14 );

true

gap> IsBijectiveOnObjects( h14 );

true

gap> IsEndomorphismWithObjects( h14 );

false

gap> IsAutomorphismWithObjects( h14 );

false

3.4 Mappings de�ned by a function

3.4.1 MappingWithObjectsByFunction

. MappingWithObjectsByFunction(src, rng, fun, imobs) (operation)

. IsMappingWithObjectsByFunction(map) (property)

. UnderlyingFunction(map) (attribute)

More general mappings, which need not preserve multiplication, are available using this operation.

See section 5.6 for an application.

Example

gap> flip := function(a) return Arrow(M78,a![1],a![3],a![2]); end;

function( a ) ... end

gap> flipmap := MappingWithObjectsByFunction( M78, M78, flip, [-8,-7] );

magma with objects mapping by function : M78 -> M78

function: function ( a )

return Arrow( M78, a![1], a![3], a![2] );

end

gap> a78; ImageElm( flipmap, a78 );

[m2 : -7 -> -8]

[m2 : -8 -> -7]



Chapter 4

Groupoids

A groupoid is a (mathematical) category in which every element is invertible. It consists of a set of

pieces, each of which is a connected groupoid. The usual terminology is `connected component', but

in GAP `component' is used for `record component', so we use the term single piece.

The simplest form for a single piece groupoid is the direct product of a group and a complete

digraph, and so is determined by a set of objects obs = W (the least of which is the root object), and

a root group grp = G. Then the elements of the groupoid are arrows g : o1 ! o2, stored as triples

[g,o1,o2], where g 2 G and o1;o2 2 W. The objects will generally be chosen to be consecutive

negative integers, but any suitable ordered set is acceptable, and `consecutive' is not a requirement.

The root group will usually be taken to be a permutation group, but pc-groups, fp-groups and matrix

groups are also supported.

A group may be considered as a single piece groupoid with one object.

A groupoid is a set of one or more single piece groupoids, its pieces, and is represented as

IsGroupoidRep, with attribute PiecesOfGroupoid.

A groupoid is homogeneous if it has two of more isomorphic pieces, with identical groups. The

special case of homogeneous, discrete groupoids, where each piece has a single object, is given its own

representation. These are used in the XMod package as the source of a crossed modules of groupoids.

For the de�nitions of the standard properties of groupoids we refer to R. Brown's book �Topology�

[Bro88], recently revised and reissued as �Topology and Groupoids� [Bro06].

4.1 Groupoids: their properties and attributes

4.1.1 SinglePieceGroupoid

. SinglePieceGroupoid(grp, obs) (operation)

. Groupoid(args) (function)

. DomainWithSingleObject(gp, obj) (operation)

The simplest construction of a groupoid is as the direct product of a group and a complete digraph.

Such a groupoid will be called a standard groupoid. Many subgroupoids of such a groupoid do not

have this simple form, and will be considered in section 4.3. The global function Groupoid will

normally �nd the appropriate constructor to call, the options being:

� the object group, a set of objects;

18

� a group being converted to a groupoid, a single object;

� a list of groupoids which have already been constructed (see 4.1.4).

Methods for ViewObj, PrintObj and Display are provided for groupoids and the other types

of object in this package. Users are advised to supply names for all the groups and groupoids they

construct.
Example

gap> s4 := Group( (1,2,3,4), (3,4) );;

gap> d8 := Subgroup( s4, [ (1,2,3,4), (1,3) ] );;

gap> SetName( s4, "s4" ); SetName( d8, "d8" );

gap> Gs4 := SinglePieceGroupoid( s4, [-15 .. -11] );

single piece groupoid: < s4, [ -15 .. -11 ] >

gap> Gd8 := Groupoid( d8, [-9,-8,-7] );

single piece groupoid: < d8, [ -9, -8, -7 ] >

gap> c6 := Group( (5,6,7)(8,9) );;

gap> SetName( c6, "c6" );

gap> Gc6 := DomainWithSingleObject( c6, -6 );

single piece groupoid: < c6, [ -6 ] >

gap> SetName( Gs4, "Gs4" ); SetName( Gd8, "Gd8" ); SetName( Gc6, "Gc6" );

4.1.2 ObjectList

. ObjectList(gpd) (attribute)

. RootObject(gpd) (attribute)

. RootGroup(gpd) (attribute)

. ObjectGroup(gpd, obj) (operation)

The ObjectList of a groupoid is the sorted list of objects. The RootObject in a single-piece

groupoid is the object with least label. A loop is an arrow of the form g : o! o, and the loops at a

particular object o form a group, the ObjectGroup at o. The RootGroup is the ObjectGroup at the

RootObject.

In the example the groupoids Gf2c6 and Gabc show that the objects need not be integers.

Example

gap> ObjectList( Gs4 );

[ -15 .. -11 ]

gap> [ RootObject( Gd8 ), RootGroup( Gc6 ), ObjectGroup( Gs4, -11 ) ];

[ -9, c6, s4 ]

gap> f2 := FreeGroup(2);;

gap> Gf2c6 := Groupoid( c6, GeneratorsOfGroup(f2) );

single piece groupoid: < c6, [ f1, f2 ] >

gap> Arrow( Gf2c6, (5,7,6), f2.1, f2.2 );

[(5,7,6) : f1 -> f2]

gap> Gabc := Groupoid( d8, [ "a", "b", "c" ] );

single piece groupoid: < d8, [ "a", "b", "c" ] >

gap> Arrow( Gabc, (2,4), "c", "b" );

[(2,4) : c -> b]

4.1.3 IsPermGroupoid

. IsPermGroupoid(gpd) (property)

. IsPcGroupoid(gpd) (property)

. IsFpGroupoid(gpd) (property)

. IsMatrixGroupoid(gpd) (property)

. IsFreeGroupoid(gpd) (property)

A groupoid is a permutation groupoid if all its pieces have permutation groups. Most of the

examples in this chapter are permutation groupoids, but in principle any type of group known toGAP

may be used.

In the following example Gf2 is an fp-groupoid and also a free groupoid, Gq8 is a pc-groupoid,

and Ggl43 is a matrix groupoid. See section 5.7 for matrix representations of groupoids.

Example

gap> f2 := FreeGroup( 2 );;

gap> Gf2 := Groupoid( f2, -22 );;

gap> SetName( f2, "f2" ); SetName( Gf2, "Gf2" );

gap> q8 := SmallGroup( 8, 4 );;

gap> Gq8 := Groupoid( q8, [ -28, -27 ] );;

gap> SetName( q8, "q8" ); SetName( Gq8, "Gq8" );

gap> gl43 := SpecialLinearGroup( 4, 3 );;

gap> Ggl43 := SinglePieceGroupoid( gl43, [ -35..-31 ] );;

gap> SetName( gl43, "gl43" ); SetName( Ggl43, "Ggl43" );

gap> [ IsMatrixGroupoid( Ggl43 ), IsFpGroupoid( Gf2 ), IsFreeGroupoid( Gf2 ),

> IsPcGroupoid( Gq8 ), IsPermGroupoid( Gs4 ) ];

[ true, true, true, true, true ]

4.1.4 UnionOfPieces

. UnionOfPieces(pieces) (operation)

. Pieces(gpd) (attribute)

. Size(gpd) (attribute)

. ReplaceOnePieceInUnion(U, old_piece, new_piece) (operation)

When a groupoid consists of two or more pieces, we require their object lists to be disjoint.

The operation UnionOfPieces and the attribute Pieces, introduced in section 2.5, are also used

for groupoids. The pieces are sorted by the least object in their object lists. The ObjectList is the

sorted concatenation of the objects in the pieces.

The Size of a groupoid is the number of its arrows. For a single piece groupoid, this is the product

of the size of the group with the square of the number of objects. For a non-connected groupoid, the

size is the sum of the sizes of its pieces.

One of the pieces in a groupoid may be replaced by an alternative piece using the operation

ReplaceOnePieceInUnion. The old_piece may be either the position of the piece to be replaced, or

one of the pieces in U. The objects in the new piece may or may not overlap the objects in the piece

being removed � we just require that the object lists in the new union are disjoint.

Example

gap> U3 := UnionOfPieces( [ Gc6, Gd8, Gs4 ] );;

gap> Display( U3 );

groupoid with 3 pieces:

< objects: [ -15 .. -11 ]

group: s4 = <[ (1,2,3,4), (3,4) ]> >

< objects: [ -9, -8, -7 ]

group: d8 = <[ (1,2,3,4), (1,3) ]> >

< objects: [ -6 ]

group: c6 = <[ (5,6,7)(8,9) ]> >

gap> Pieces( U3 );

[ Gs4, Gd8, Gc6 ]

gap> ObjectList( U3 );

[ -15, -14, -13, -12, -11, -9, -8, -7, -6 ]

gap> U2 := Groupoid( [ Gf2, Gq8 ] );;

gap> [ Size(Gs4), Size(Gd8), Size(Gc6), Size(U3) ];

[ 600, 72, 6, 678 ]

gap> [ Size(Gf2), Size(Gq8), Size(U2) ];

[ infinity, 32, infinity ]

gap> U5 := UnionOfPieces( [ U3, U2 ] );

groupoid with 5 pieces:

[ Gq8, Gf2, Gs4, Gd8, Gc6 ]

gap> Display( U5 );

groupoid with 5 pieces:

< objects: [ -28, -27 ]

group: q8 = <[ f1, f2, f3 ]> >

< objects: [ -22 ]

group: f2 = <[ f1, f2 ]> >

< objects: [ -15 .. -11 ]

group: s4 = <[ (1,2,3,4), (3,4) ]> >

< objects: [ -9, -8, -7 ]

group: d8 = <[ (1,2,3,4), (1,3) ]> >

< objects: [ -6 ]

group: c6 = <[ (5,6,7)(8,9) ]> >

gap> V5 := ReplaceOnePieceInUnion( U5, 3, Ggl43 );

groupoid with 5 pieces:

[ Ggl43, Gq8, Gf2, Gd8, Gc6 ]

gap> ObjectList( V5 );

[ -35, -34, -33, -32, -31, -28, -27, -22, -9, -8, -7, -6 ]

gap> U5 = V5;

false

gap> W5 := ReplaceOnePieceInUnion( V5, Gc6, Gs4 );

groupoid with 5 pieces:

[ Ggl43, Gq8, Gf2, Gs4, Gd8 ]

gap> ObjectList( W5 );

[ -35, -34, -33, -32, -31, -28, -27, -22, -15, -14, -13, -12, -11, -9, -8, -7

]

4.1.5 HomogeneousGroupoid

. HomogeneousGroupoid(gpd, oblist) (operation)

. HomogeneousDiscreteGroupoid(gp, obs) (operation)

Special functions are provided for the case where a groupoid has more than one connected com-

ponent, and when these components are identical except for their object sets. Such groupoids are said

to be homogeneous.

The operation HomogeneousGroupoid is used when the components each contain more than one

object. The arguments consist of a single piece groupoid gpd and a list of lists of objects oblist,

each of whose lists has the same length as the object list obs of gpd. Note that gpd is not included as

one of the pieces in the output unless obs is included as one of the lists in oblist.

The operation HomogeneousDiscreteGroupoid is used when the components each have a single

object. In this case the �rst argument is just a group � the root group for each component. These

groupoids are used in the XMod package as the source of many crossed modules of groupoids.

Both types of groupoid have the property IsHomogeneousDomainWithObjects. In the latter case

a separate representation IsHomogeneousDiscreteGroupoidRep is used.

Example

gap> Hd8 := HomogeneousGroupoid( Gd8, [ [-12,-11,-10], [-16,-15,-14] ] );

homogeneous groupoid with 2 pieces:

1: single piece groupoid: < d8, [ -16, -15, -14 ] >

2: single piece groupoid: < d8, [ -12, -11, -10 ] >

gap> Size(Hd8); ## 8x3x3 + 8x3x3

144

gap> IsHomogeneousDomainWithObjects( Hd8 );

true

gap> Hc6 := HomogeneousDiscreteGroupoid( c6, [-7..-4] );

homogeneous, discrete groupoid: < c6, [ -7 .. -4 ] >

gap> Size(Hc6); ## 6x4

24

gap> RepresentationsOfObject(Gd8);

[ "IsComponentObjectRep", "IsAttributeStoringRep", "IsMWOSinglePieceRep" ]

gap> RepresentationsOfObject(Hd8);

[ "IsComponentObjectRep", "IsAttributeStoringRep", "IsPiecesRep" ]

gap> RepresentationsOfObject(Hc6);

[ "IsComponentObjectRep", "IsAttributeStoringRep",

"IsHomogeneousDiscreteGroupoidRep" ]

gap> ktpo := KnownTruePropertiesOfObject(Hc6);;

gap> ans :=

> [ "CanEasilyCompareElements", "CanEasilySortElements",

> "IsDuplicateFree", "IsAssociative", "IsCommutative",

> "IsDiscreteDomainWithObjects", "IsHomogeneousDomainWithObjects" ];;

gap> ForAll( ans, a -> ( a in ktpo ) );

true

4.1.6 DirectProductOp

. DirectProductOp(list, gpd) (operation)

. Projection(gpd, pos) (operation)

. Embedding(gpd, pos) (operation)

The direct product of groupoids G;H has as group the direct product of the groups in G and H

and as object list the cartesian product of their object lists.

Example

gap> prod := DirectProductOp( [Gd8,Gc6], Gd8 );

single piece groupoid: < Group( [ (1,2,3,4), (1,3), (5,6,7)(8,9) ] ),

[ [ -9, -6 ], [ -8, -6 ], [ -7, -6 ] ] >

gap> Projection( prod, 1 );

groupoid homomorphism :

[ [ [(1,2,3,4) : [ -9, -6 ] -> [ -9, -6 ]], [(1,3) : [ -9, -6 ] -> [ -9, -6 ]]

, [(5,6,7)(8,9) : [ -9, -6 ] -> [ -9, -6 ]],

[() : [ -9, -6 ] -> [ -8, -6 ]], [() : [ -9, -6 ] -> [ -7, -6 ]] ],

[ [(1,2,3,4) : -9 -> -9], [(1,3) : -9 -> -9], [() : -9 -> -9],

[() : -9 -> -8], [() : -9 -> -7] ] ]

gap> Embedding(prod,2);

groupoid homomorphism :

[ [ [(5,6,7)(8,9) : -6 -> -6] ],

[ [(5,6,7)(8,9) : [ -9, -6 ] -> [ -9, -6 ]] ] ]

gap> DirectProductInfo( prod );

rec( embeddings := [ , groupoid homomorphism :

[ [ [(5,6,7)(8,9) : -6 -> -6] ],

[ [(5,6,7)(8,9) : [ -9, -6 ] -> [ -9, -6 ]] ] ] ], first := Gd8,

groupoids := [ Gd8, Gc6 ], groups := [ d8, c6 ],

objectlists := [ [ -9, -8, -7 ], [ -6 ] ],

projections := [ groupoid homomorphism :

[ [ [(1,2,3,4) : [ -9, -6 ] -> [ -9, -6 ]],

[(1,3) : [ -9, -6 ] -> [ -9, -6 ]],

[(5,6,7)(8,9) : [ -9, -6 ] -> [ -9, -6 ]],

[() : [ -9, -6 ] -> [ -8, -6 ]],

[() : [ -9, -6 ] -> [ -7, -6 ]] ],

[ [(1,2,3,4) : -9 -> -9], [(1,3) : -9 -> -9], [() : -9 -> -9],

[() : -9 -> -8], [() : -9 -> -7] ] ] ] )

4.2 Groupoid elements; stars; costars; homsets

4.2.1 GroupoidElement

. GroupoidElement(gpd, elt, tail, head) (operation)

. ElementOfArrow(elt) (operation)

. TailOfArrow(elt) (operation)

. HeadOfArrow(elt) (operation)

The operation GroupoidElement is a synonym for the operation Arrow, as described in subsec-

tion 2.1.2. To recapitulate, an arrow e consists of a group element, ElementOfArrow(e); the tail

(source) object, TailOfArrow(e); and the head (target) object, HeadOfArrow(e). Arrows have a

partial composition: two arrows may be multiplied when the head of the �rst coincides with the tail

of the second. If an attempt is made to multiply arrows where this condition does not hold, then the

value fail is returned.

Example

gap> e1 := GroupoidElement( Gd8, (1,2,3,4), -9, -8 );

[(1,2,3,4) : -9 -> -8]

gap> e2 := Arrow( Gd8, (1,3), -8, -7 );

[(1,3) : -8 -> -7]

gap> Print( [ ElementOfArrow(e1), TailOfArrow(e1), HeadOfArrow(e1) ], "\n" );

[ (1,2,3,4), -9, -8 ]

gap> prod := e1*e2;

[(1,2)(3,4) : -9 -> -7]

gap> e2*e1;

fail

gap> e3 := Arrow( Gd8, (2,4), -7, -9 );;

gap> loop := prod*e3;

[(1,4,3,2) : -9 -> -9]

gap> loop^2;

[(1,3)(2,4) : -9 -> -9]

4.2.2 IdentityArrow

. IdentityArrow(gpd, obj) (operation)

The identity arrow 1o of G at object o is (e : o! o) where e is the identity element in the object

group. The inverse arrow e�1 of e= (c : p! q) is (c�1 : q! p), so that e�e�1 = 1p and e
�1 �e= 1q.

Example

gap> i8 := IdentityArrow( Gd8, -8 );

[() : -8 -> -8]

gap> [ e1*i8, i8*e1, e1^-1];

[ [(1,2,3,4) : -9 -> -8], fail, [(1,4,3,2) : -8 -> -9] ]

4.2.3 Order

. Order(arr) (attribute)

A groupoid element is a loop when the tail and head coincide. In this case the order of the element

is de�ned to be the order of its group element.

Example

gap> i8; Order(i8);

[() : -8 -> -8]

1

gap> loop; Order(loop);

[(1,4,3,2) : -9 -> -9]

4

4.2.4 ObjectStar

. ObjectStar(gpd, obj) (operation)

. ObjectCostar(gpd, obj) (operation)

. Homset(gpd, tail, head) (operation)

The star at obj is the set of arrows which have obj as tail, while the costar is the set of arrows

which have obj as head. The homset from obj1 to obj2 is the set of arrows with the speci�ed tail

and head, and so is bijective with the elements of the object groups. Thus every star and every costar

is a union of homsets. The identity arrow at an object is a left identity for the star and a right identity

for the costar at that object.

In order not to create unneccessarily long lists, these operations return objects of type

IsHomsetCosetsRep for which an Iterator is provided. (An Enumerator is not yet implemented.)

Example

gap> star9 := ObjectStar( Gd8, -9 );

<star at [ -9 ] with group d8>

gap> Size( star9 );

24

gap> for e in star9 do

> if ( Order( ElementOfArrow(e) ) = 4 ) then Print( e, "\n" ); fi;

> od;

[(1,4,3,2) : -9 -> -9]

[(1,4,3,2) : -9 -> -8]

[(1,4,3,2) : -9 -> -7]

[(1,2,3,4) : -9 -> -9]

[(1,2,3,4) : -9 -> -8]

[(1,2,3,4) : -9 -> -7]

gap> costar6 := ObjectCostar( Gc6, -6 );

<costar at [ -6 ] with group c6>

gap> Size( costar6 );

6

gap> hsetq8 := Homset( Gq8, -28, -27 );

<homset -28 -> -27 with group q8>

gap> for e in hsetq8 do Print(e,"\n"); od;

[<identity> of ... : -28 -> -27]

[f3 : -28 -> -27]

[f2 : -28 -> -27]

[f2*f3 : -28 -> -27]

[f1 : -28 -> -27]

[f1*f3 : -28 -> -27]

[f1*f2 : -28 -> -27]

[f1*f2*f3 : -28 -> -27]

4.3 Subgroupoids

4.3.1 Subgroupoid

. Subgroupoid(args) (function)

. SubgroupoidByPieces(gpd, obhoms) (operation)

. IsSubgroupoid(gpd, sgpd) (operation)

. IsWideSubgroupoid(gpd, sgpd) (operation)

A subgroupoid S of a groupoid G has as objects some subset of the objects of G. It is wide in G if

both groupoids have the same object set. It is full if, for any two objects in S, the Homset is the same

as that in G. The arrows of S are a subset of those of G, closed under multiplication and with tail and

head in the chosen object set.

There are a variety of constructors for a subgroupoid of a standard groupoid, and the most general

is the operation SubgroupoidByPieces. Its two parameters are a groupoid and a list of pieces, each

piece being speci�ed as a list [sgp,obs], where sgp is a subgroup of the root group in that piece,

and obs is a subset of the objects in that piece.

Example

gap> c4 := Subgroup( d8, [ (1,2,3,4) ] );;

gap> k4 := Subgroup( d8, [ (1,2)(3,4), (1,3)(2,4) ] );;

gap> SetName( c4, "c4" ); SetName( k4, "k4" );

gap> Ud8 := Subgroupoid( Gd8, [ [ k4, [-9] ], [ c4, [-8,-7] ] ] );;

gap> SetName( Ud8, "Ud8" );

gap> Display( Ud8 );

groupoid with 2 pieces:

< objects: [ -9 ]

group: k4 = <[ (1,2)(3,4), (1,3)(2,4) ]> >

< objects: [ -8, -7 ]

group: c4 = <[ (1,2,3,4) ]> >

gap> [ Parent( Ud8 ), IsWide( Gd8, Ud8 ) ];

[ Gd8, true ]

gap> genf2b := List( GeneratorsOfGroup(f2), g -> g^2 );

[ f1^2, f2^2 ]

gap> f2b := Subgroup( f2, genf2b );;

gap> SU2 := SubgroupoidByPieces( U2, [ [q8,[-27]], [f2b,[-22]] ] );

groupoid with 2 pieces:

1: single piece groupoid: < q8, [ -27 ] >

2: single piece groupoid: < Group( [ f1^2, f2^2 ] ), [ -22 ] >

gap> IsWideSubgroupoid( U2, SU2 );

false

gap> IsSubgroupoid( Gf2, Groupoid( f2b, [-22] ) );

true

4.3.2 FullSubgroupoid

. FullSubgroupoid(gpd, obs) (operation)

. FullTrivialSubgroupoid(gpd) (attribute)

The FullSubgroupoid of a groupoid gpd on a subset obs of its objects contains all the arrows

of gpd with tail and head in obs. A trivial subgroupoid has trivial object groups, but need not be

discrete. A single piece trivial groupoid is sometimes called a tree groupoid. (The term identity

subgroupoid was used in versions up to 1.14.) The global function Subgroupoid should call the

appropriate operation.

Example

gap> FullSubgroupoid( U3, [-7,-6] );

groupoid with 2 pieces:

1: single piece groupoid: < d8, [ -7 ] >

2: single piece groupoid: < c6, [ -6 ] >

gap> FullTrivialSubgroupoid( Ud8 );

groupoid with 2 pieces:

1: single piece groupoid: < id(k4), [ -9 ] >

2: single piece groupoid: < id(c4), [ -8, -7 ] >

4.3.3 DiscreteSubgroupoid

. DiscreteSubgroupoid(gpd, sgps, obs) (operation)

. MaximalDiscreteSubgroupoid(gpd) (attribute)

. DiscreteTrivialSubgroupoid(gpd) (attribute)

A subgroupoid is discrete if it is a union of groups. The MaximalDiscreteSubgroupoid of gpd

is the union of all the single-object full subgroupoids of gpd.

Example

gap> DiscreteSubgroupoid( U3, [ c4, k4 ], [-9,-7] );

groupoid with 2 pieces:

1: single piece groupoid: < c4, [ -9 ] >

2: single piece groupoid: < k4, [ -7 ] >

gap> MaximalDiscreteSubgroupoid(U2);

groupoid with 3 pieces:

1: single piece groupoid: < q8, [ -28 ] >

2: single piece groupoid: < q8, [ -27 ] >

3: single piece groupoid: < f2, [ -22 ] >

gap> DiscreteTrivialSubgroupoid( U2 );

groupoid with 3 pieces:

1: single piece groupoid: < id(q8), [ -28 ] >

2: single piece groupoid: < id(q8), [ -27 ] >

3: single piece groupoid: < id(f2), [ -22 ] >

4.3.4 SubgroupoidWithRays

. SubgroupoidWithRays(gpd, sgp, rays) (operation)

. RaysOfGroupoid(gpd) (operation)

. RayElementsOfGroupoid(gpd) (operation)

If groupoid G is of type IsDirectProductWithCompleteDigraph with group g and n objects,

then a typical wide subgroupoid H of G is formed by choosing a subgroup h of g to be the object

group at the root object q, and an arrow r : q! p for each of the objects p. The chosen loop arrow

at q must be the identity arrow. These n arrows are called the rays of the subgroupoid. The arrows in

the homset from p to p0 have the form r�1xr0 where r;r0 are the rays from q to p; p0 respectively, and

x 2 h.

The operation RaysOfGroupoid returns a list of arrows, one for each object, while the operation

RayElementsOfGroupoid returns the list of group elements in these arrows.

In the following example we construct a subgroupoid with rays on three of the �ve objects. It is

therefore necessary to construct the full subgroupoid on these three objects �rst.

Note that it is also possible to construct a subgroupoid with rays of a subgroupoid with rays.

Note also that the function Ancestor provides an iteration of Parent.

Example

gap> Hs4 := FullSubgroupoid( Gs4, [-14,-13,-12] );;

gap> SetName( Hs4, "Hs4" );

gap> Hd8a := SubgroupoidWithRays( Hs4, d8, [(),(2,3),(3,4)] );

single piece groupoid with rays: < d8, [ -14, -13, -12 ], [ (), (2,3), (3,4)

] >

gap> hs1413 := Homset( Hd8a, -14, -13 );

<homset -14 -> -13 with group d8>

gap> for e in hs1413 do Print(e,", "); od; Print( "\n");

[(2,3) : -14 -> -13], [(2,4,3) : -14 -> -13], [(1,2,4,3) : -14 -> -13], [

(1,2,3) : -14 -> -13], [(1,4,2) : -14 -> -13], [(1,4) : -14 -> -13], [

(1,3,4) : -14 -> -13], [(1,3,4,2) : -14 -> -13],

gap> Hd8b := SubgroupoidWithRays( Hs4, d8, [(),(1,2,3),(1,2,4)] );

single piece groupoid with rays: < d8, [ -14, -13, -12 ],

[ (), (1,2,3), (1,2,4) ] >

gap> Hd8a = Hd8b;

true

gap> SetName( Hd8b, "Hd8b" );

gap> RayElementsOfGroupoid( Hd8b );

[ (), (1,2,3), (1,2,4) ]

gap> Parent( Hd8a );

Hs4

gap> Ancestor( Hd8a );

Gs4

gap> Fd8a := FullSubgroupoid( Hd8a, [-13,-12] );

single piece groupoid with rays: < Group( [ (1,3,2,4), (1,2) ] ),

[ -13, -12 ], [ (), (2,4,3) ] >

gap> Kd8a := SubgroupoidWithRays( Fd8a, k4, [ (), (1,3) ] );

single piece groupoid with rays: < k4, [ -13, -12 ], [ (), (1,3) ] >

4.3.5 SinglePieceSubgroupoidByGenerators

. SinglePieceSubgroupoidByGenerators(parent, gens) (operation)

A set of arrows generates a groupoid by taking all possible products and inverses. So far, the only

implementation is for the case of loops generating a group at an object o together with a set of rays

from o, where o is not the least object. A suitably large supergroupoid, which must be a direct product

with a complete digraph, should be provided. This is the case needed for ConjugateGroupoid in

section 4.5.2. Other cases will be added as time permits.
Example

gap> u := Arrow( Gs4, (1,2,3), -15, -13 );

[(1,2,3) : -15 -> -13]

gap> gensa := GeneratorsOfGroupoid( Hd8a );

[ [(1,2,3,4) : -14 -> -14], [(1,3) : -14 -> -14], [(2,3) : -14 -> -13],

[(3,4) : -14 -> -12] ]

gap> imsa := List( gensa, g -> g^u );

[ [(1,2,3,4) : -14 -> -14], [(1,3) : -14 -> -14], [(1,3) : -14 -> -15],

[(3,4) : -14 -> -12] ]

gap> C := SinglePieceSubgroupoidByGenerators( Gs4, imsa );

single piece groupoid with rays: < Group( [ (1,4,3,2), (1,3) ] ),

[ -15, -14, -12 ], [ (), (1,3), (1,4,3) ] >

4.4 Left, right and double cosets

4.4.1 RightCoset

. RightCoset(G, U, elt) (operation)

. RightCosetRepresentatives(G, U) (operation)

. LeftCoset(G, U, elt) (operation)

. LeftCosetRepresentatives(G, U) (operation)

. LeftCosetRepresentativesFromObject(G, U, obj) (operation)

. DoubleCoset(G, U, elt, V) (operation)

. DoubleCosetRepresentatives(G, U, V) (operation)

If U is a wide subgroupoid of G, the right cosets Ug ofU in G are the equivalence classes for the

relation on the arrows of G where g1 is related to g2 if and only if g2 = u � g1 for some arrow u of

U . The right coset containing g is written Ug. These right cosets partition the costars of G and, in

particular, the costarU1_o ofU at object o, so that (unlike groups)U is itself a coset only when G has

a single object.

The right coset representatives forU in G form a list containing one arrow for each coset where,

in a particular piece ofU , the group element chosen is the right coset representative of the group ofU

in the group of G.

Similarly, the left cosets gU re�ne the stars of G, while double cosets are unions of left cosets

and right cosets. The operation LeftCosetRepresentativesFromObject( G, U, obj ) is used

in Chapter 4, and returns only those representatives which have tail at obj.

As with stars and homsets, these cosets are implemented with representation

IsHomsetCosetsRep and provided with an iterator. Note that, when U has more than one

piece, cosets may have differing lengths.

In the example the representative for the right coset re2 is the last one in the list rcrd8, namely

[(2,4):-8->-7].
Example

gap> re2 := RightCoset( Gd8, Ud8, e2 );

RightCoset(single piece groupoid: < c4, [ -8, -7 ] >,[(1,3) : -8 -> -7])

gap> for x in re2 do Print( x, "\n" ); od;

[(1,3) : -8 -> -7]

[(1,3) : -7 -> -7]

[(2,4) : -8 -> -7]

[(2,4) : -7 -> -7]

[(1,4)(2,3) : -8 -> -7]

[(1,4)(2,3) : -7 -> -7]

[(1,2)(3,4) : -8 -> -7]

[(1,2)(3,4) : -7 -> -7]

gap> rcrd8 := RightCosetRepresentatives( Gd8, Ud8 );

[ [() : -9 -> -9], [() : -9 -> -8], [() : -9 -> -7], [(2,4) : -9 -> -9],

[(2,4) : -9 -> -8], [(2,4) : -9 -> -7], [() : -8 -> -9], [() : -8 -> -8],

[() : -8 -> -7], [(2,4) : -8 -> -9], [(2,4) : -8 -> -8], [(2,4) : -8 -> -7]

]

gap> lcr7 := LeftCosetRepresentativesFromObject( Gd8, Ud8, -7 );

[ [() : -7 -> -9], [(2,4) : -7 -> -9], [() : -7 -> -8], [(2,4) : -7 -> -8] ]

4.5 Conjugation

4.5.1 \^

. \^(e1, e) (operation)

Conjugation by an arrow e = (c : p! q) is the groupoid inner automorphism (see section 5.6)

de�ned as follows. There are two cases to consider. In the case p 6= q,

� objects p;q are interchanged, and the remaining objects are �xed;

� loops at p;q: (b : p! p) 7! (bc : q! q) and (b : q! q) 7! (bc
�1

: p! p);

� arrows between p and q: (b : p! q) 7! (c�1bc�1 : q! p) and (b : q! p) 7! (cbc : p! q);

� costars at p;q: (b : r! p) 7! (bc : r! q) and (b : r! q) 7! (bc�1 : r! p);

� stars at p;q: (b : p! r) 7! (c�1b :! q) and (b : q! r) 7! (cb : p! r);

� the remaining arrows are unchanged.

In the case p= q,

� all the objects are �xed;

� loops at p are conjugated by c, so (b : p! p) 7! (bc : p! p);

� the rest of the costar and star at p are permuted,

(b : r! p) 7! (bc : r! p) and (b : p! r) 7! (c�1b : p! r);

� the remaining arrows are unchanged.

The details of this construction may be found in section 3.2 of [AW10].

Example

gap> x := Arrow( Gd8, (1,3), -9, -9 );;

gap> y := Arrow( Gd8, (1,2,3,4), -8, -9 );;

gap> z := Arrow( Gd8, (1,2)(3,4), -9, -7 );;

gap> w := Arrow( Gd8, (1,2,3,4), -7, -8 );;

gap> ## conjugation with elements x, y, and z in Gd8:

gap> x^y;

[(2,4) : -8 -> -8]

gap> x^z;

[(2,4) : -7 -> -7]

gap> y^x;

[() : -8 -> -9]

gap> y^z;

[(2,4) : -8 -> -7]

gap> z^x;

[(1,4,3,2) : -9 -> -7]

gap> z^y;

[(2,4) : -8 -> -7]

gap> w^z

[(1,3) : -9 -> -8]

4.5.2 ConjugateGroupoid

. ConjugateGroupoid(gpd, e) (operation)

When H is a subgroupoid of a groupoid G and a is an arrow of G, then the conjugate of H by a is

the subgroupoid generated by the conjugates of the generators of H.

Example

gap> ConjugateGroupoid( Hd8a, u^-1 );

single piece groupoid with rays: < Group( [ (1,4,3,2), (1,3) ] ),

[ -15, -14, -12 ], [ (), (1,3), (1,4,3) ] >

More examples of all these operations may be found in the example �le

groupoids/examples/gpd.g.



Chapter 5

Homomorphisms of Groupoids

A homomorphism m from a groupoid G to a groupoid H consists of a map from the objects of G to

those of H together with a map from the elements of G to those of H which is compatible with tail

and head and which preserves multiplication:

m(g1 : o1! o2)�m(g2 : o2! o3) = m(g1�g2 : o1! o3):

Note that when a homomorphism is not injective on objects, the image of the source need not be a

subgroupoid of the range. A simple example of this is given by a homomorphism from the two-object,

four-element groupoid with trivial group to the free group hai on one generator, when the image is

[1;an;a�n] for some n> 0.

A variety of homomorphism operations are available.

� The basic construction is a homomorphism f : G! H from a connected groupoid G to a con-

nected groupoid H, constructed using GroupoidHomomorphismFromSinglePiece, (see 5.1).

� Since more than one connected groupoid may be mapped to the same range, we then have the

operation GroupoidHomomorphismToSinglePiece, (see 5.4).

� The third case arises when the range is a union of connected groupoids, in which case

HomomorphismByUnion is called, (see 5.5).

� Fourthly, there are is an additional operation for the case where the source is homogeneous and

discrete, GroupoidHomomorphismFromHomogeneousDiscrete, (see 5.4.2).

� Finally, there are special operations for inclusion mappings, restriced mappings, and groupoid

automorphisms (see 5.6).

5.1 Homomorphisms from a connected groupoid

5.1.1 GroupoidHomomorphismFromSinglePiece

. GroupoidHomomorphismFromSinglePiece(src, rng, gens, images) (operation)

. GroupoidHomomorphism(args) (function)

. MappingToSinglePieceData(map) (operation)

32

The simplest groupoid homomorphism is a mapping f : G! H from a connected groupoid G to

a connected groupoid H. There are two equivalent sets of input data which may be used. Both require

the Source G and the Range H. The �rst then requires:

� the set of generating arrows GeneratorsOfGroupoid(G);

� a list of image arrows in H.

This data is stored in the attribute MappingGeneratorsImages.

The alternative input data consists of:

� a homomorphism rhom from the root group of G to the group at the image object in H;

� a list imobs of the images of the objects of G;

� a list imrays of the elements in the images of the rays of G, so that the image f(ri : o1 ! oi) of
the i-th ray is (imrays[i]:imobs[1]!imobs[i]).

This data is stored in the attribute MappingToSinglePieceData.

So an alternative way to construct a homomorphism of groupoids is to make a call of the form

GroupoidHomomorphism(src,rng,rhom,imobs,imrays).

In the following example the same homomorphism is constructed using both methods.

Example

gap> gen1 := GeneratorsOfGroupoid( Gq8 );

[ [f1 : -28 -> -28], [f2 : -28 -> -28], [f3 : -28 -> -28],

[<identity> of ... : -28 -> -27] ]

gap> gen2 := GeneratorsOfGroupoid( Hd8b );

[ [(1,2,3,4) : -14 -> -14], [(1,3) : -14 -> -14], [(1,2,3) : -14 -> -13],

[(1,2,4) : -14 -> -12] ]

gap> images := [ gen2[1]^2, gen2[1]*gen2[2], IdentityArrow(Hd8b,-14), gen2[4] ];

[ [(1,3)(2,4) : -14 -> -14], [(1,2)(3,4) : -14 -> -14], [() : -14 -> -14],

[(1,2,4) : -14 -> -12] ]

gap> mor1 := GroupoidHomomorphism( Gq8, Hd8b, gen1, images );

groupoid homomorphism : Gq8 -> Hd8b

[ [ [f1 : -28 -> -28], [f2 : -28 -> -28], [f3 : -28 -> -28],

[<identity> of ... : -28 -> -27] ],

[ [(1,3)(2,4) : -14 -> -14], [(1,2)(3,4) : -14 -> -14], [() : -14 -> -14],

[(1,2,4) : -14 -> -12] ] ]

gap> genq8 := GeneratorsOfGroup( q8 );;

gap> imh := [ (1,3)(2,4), (1,2)(3,4), () ];;

gap> h := GroupHomomorphismByImages( q8, d8, genq8, imh );

[ f1, f2, f3 ] -> [ (1,3)(2,4), (1,2)(3,4), () ]

gap> mor2 := GroupoidHomomorphism( Gq8, Hd8b, h, [-14,-12], [(),(1,2,4)] );;

gap> mor1=mor2;

true

gap> e := Arrow( Gq8, Product(genq8), -27, -28 );

[f1*f2*f3 : -27 -> -28]

gap> ImageElm( mor2, e );

[(2,4,3) : -12 -> -14]

5.2 Properties and attributes of groupoid homomorphisms

5.2.1 Properties of a groupoid homomorphism

The properties listed in subsection 3.3 for homomorphisms of magmas with objects also apply to

groupoid homomorphisms.
Example

gap> IsInjectiveOnObjects( mor2 );

true

gap> IsSurjectiveOnObjects( mor2 );

false

gap> ad8 := GroupHomomorphismByImages( d8, d8,

> [ (1,2,3,4), (1,3) ], [ (1,4,3,2), (2,4) ] );;

gap> md8 := GroupoidHomomorphism( Gd8, Gd8, ad8, [-7,-9,-8], [(),(1,3),(2,4)] );

groupoid homomorphism : Gd8 -> Gd8

[ [ [(1,2,3,4) : -9 -> -9], [(1,3) : -9 -> -9], [() : -9 -> -8],

[() : -9 -> -7] ],

[ [(1,4,3,2) : -7 -> -7], [(2,4) : -7 -> -7], [(1,3) : -7 -> -9],

[(2,4) : -7 -> -8] ] ]

gap> IsBijectiveOnObjects( md8 );

true

gap> IsEndomorphismWithObjects( md8 );

true

gap> IsAutomorphismWithObjects( md8 );

true

5.2.2 Attributes of a groupoid homomorphism

The attributes of a groupoid homomorphism mor from a single piece groupoid cover both forms of

construction de�ned above.

� S = Source(mor) is the source groupoid of the homomorphism;

� R = Range(mor) is the range groupoid of the homomorphism;

� RootGroupHomomorphism(mor) is the group homomorphism from the root group of S to the

group at the image object in R of the root object in S;

� ImagesOfObjects(mor) is the list of objects in R which are the images of the objects in S;

� ImageElementsOfRays(mor) is the list of group elements in those arrows in R which are the

images of the rays in S;

� MappingGeneratorsImages(mor) is the two element list containing the list of generators in

S and the list of their images in R;

� MappingToSinglePieceData(mor) is a list with three elements: the root group homomor-

phism; the images of the objects; and the images of the rays.

For other types of homomorphism the attributes are very similar.

The operation ObjectGroupHomomorphism, though an operation, is included in this section for

convenience.

5.2.3 RootGroupHomomorphism

. RootGroupHomomorphism(hom) (attribute)

This is the group homomorphism from the root group of the source groupoid to the group at the

image object in the range groupoid of the root object in the source.

5.2.4 ImagesOfObjects

. ImagesOfObjects(hom) (attribute)

This is the list of objects in the range groupoid which are the images of the objects in the source.

5.2.5 ImageElementsOfRays

. ImageElementsOfRays(hom) (attribute)

This is the list of group elements in those arrows in the range groupoid which are the images of

the rays in the source.

Example

gap> RootGroupHomomorphism( mor2 );

[ f1, f2, f3 ] -> [ (1,3)(2,4), (1,2)(3,4), () ]

gap> ImagesOfObjects( mor2 );

[ -14, -12 ]

gap> ImageElementsOfRays( mor2 );

[ (), (1,2,4) ]

5.2.6 ObjectGroupHomomorphism

. ObjectGroupHomomorphism(gpdhom, obj) (operation)

This operations gives the group homomorphism from an object group of the source to the object

group at the image object in the range.

Example

gap> ObjectGroupHomomorphism( mor1, -27 );

[ f1, f2, f3 ] -> [ (1,4)(2,3), (1,3)(2,4), () ]

5.3 Special types of groupoid homomorphism

In this section we mention inclusion mappings of subgroupoids; and mappings restricted to a source

subgroupoid. We also discuss various types of isomorphism: to a different set of objects; to a permu-

tation groupoid; to a pc-groupoid.

5.3.1 InclusionMappingGroupoids

. InclusionMappingGroupoids(gpd, sgpd) (operation)

The operation InclusionMappingGroupoids(gpd,sgpd) returns the inclusion homomorphism

from the subgroupoid sgpd to gpd.

Example

gap> inc := InclusionMappingGroupoids( Hs4, Hd8b );

groupoid homomorphism : Hd8b -> Hs4

[ [ [(1,2,3,4) : -14 -> -14], [(1,3) : -14 -> -14], [(1,2,3) : -14 -> -13],

[(1,2,4) : -14 -> -12] ],

[ [(1,2,3,4) : -14 -> -14], [(1,3) : -14 -> -14], [(1,2,3) : -14 -> -13],

[(1,2,4) : -14 -> -12] ] ]

5.3.2 RestrictedMappingGroupoids

. RestrictedMappingGroupoids(mor, sgpd) (operation)

The operation RestrictedMappingGroupoids(mor,sgpd) returns the restriction of the homo-

morphism mor to the subgroupoid sgpd of its source. The range is usually set to the ImagesSource

of the restriction. For another example see section 5.7

Example

gap> max := MaximalDiscreteSubgroupoid( Hd8b );;

gap> res := RestrictedMappingGroupoids( inc, max );

groupoid homomorphism from several pieces :

groupoid homomorphism :

[ [ [(1,2,3,4) : -14 -> -14], [(1,3) : -14 -> -14] ],

[ [(1,2,3,4) : -14 -> -14], [(1,3) : -14 -> -14] ] ]

groupoid homomorphism :

[ [ [(1,4,2,3) : -13 -> -13], [(1,2) : -13 -> -13] ],

[ [(1,4,2,3) : -13 -> -13], [(1,2) : -13 -> -13] ] ]

groupoid homomorphism :

[ [ [(1,2,4,3) : -12 -> -12], [(2,3) : -12 -> -12] ],

[ [(1,2,4,3) : -12 -> -12], [(2,3) : -12 -> -12] ] ]

5.3.3 IsomorphismNewObjects

. IsomorphismNewObjects(src, objlist) (operation)

. IsomorphismStandardGroupoid(gpd, obs) (operation)

The operation IsomorphismNewObjects(gpd,obs) returns the isomorphism from a groupoid

gpd to a groupoid with the same object group and ray elements but with a different set obs of objects.

The operation IsomorphismStandardGroupoid(gpd,obs) returns the isomorphism from a

groupoid with rays to the groupoid of type IsDirectProductWithCompleteDigraphDomain on

the given set obs of objects.



groupoids 37

Example

gap> iso1 := IsomorphismNewObjects( Hs4, [-30,-20,-10] );

groupoid homomorphism :

[ [ [(1,2,3,4) : -14 -> -14], [(3,4) : -14 -> -14], [() : -14 -> -13],

[() : -14 -> -12] ],

[ [(1,2,3,4) : -30 -> -30], [(3,4) : -30 -> -30], [() : -30 -> -20],

[() : -30 -> -10] ] ]

gap> inc2 := mor2*inc*iso1;

groupoid homomorphism :

[ [ [f1 : -28 -> -28], [f2 : -28 -> -28], [f3 : -28 -> -28],

[<identity> of ... : -28 -> -27] ],

[ [(1,3)(2,4) : -30 -> -30], [(1,2)(3,4) : -30 -> -30], [() : -30 -> -30],

[(1,2,4) : -30 -> -10] ] ]

gap> iso2 := IsomorphismStandardGroupoid( Hd8b, [-23,-22,-21] );

groupoid homomorphism :

[ [ [(1,2,3,4) : -14 -> -14], [(1,3) : -14 -> -14], [(1,2,3) : -14 -> -13],

[(1,2,4) : -14 -> -12] ],

[ [(1,2,3,4) : -23 -> -23], [(1,3) : -23 -> -23], [() : -23 -> -22],

[() : -23 -> -21] ] ]

gap> inv2 := InverseGeneralMapping( iso2 );

groupoid homomorphism :

[ [ [(1,2,3,4) : -23 -> -23], [(1,3) : -23 -> -23], [() : -23 -> -22],

[() : -23 -> -21] ],

[ [(1,2,3,4) : -14 -> -14], [(1,3) : -14 -> -14], [(1,2,3) : -14 -> -13],

[(1,2,4) : -14 -> -12] ] ]

5.3.4 IsomorphismPermGroupoid

. IsomorphismPermGroupoid(gpd) (attribute)

. IsomorphismPcGroupoid(gpd) (operation)

The operation IsomorphismPermGroupoid(gpd) returns an isomorphism from a groupoid

gpd to a groupoid with the same objects but with an isomorphic permutation group. Similarly,

IsomorphismPcGroupoid(gpd) changes the group into a pc-group (if appropriate).

Example

gap> N2 := NormalSubgroups( q8 )[2];;

gap> Hq8 := SubgroupoidWithRays( Gq8, N2, [ One(q8), q8.1 ] );

single piece groupoid with rays: < Group( [ f2, f3 ] ), [ -28, -27 ],

[ <identity> of ..., f1 ] >

gap> isoHq8 := IsomorphismPermGroupoid( Hq8 );

groupoid homomorphism :

[ [ [f2 : -28 -> -28], [f3 : -28 -> -28], [f1 : -28 -> -27] ],

[ [(1,3,4,7)(2,5,6,8) : -28 -> -28], [(1,4)(2,6)(3,7)(5,8) : -28 -> -28],

[(1,2,4,6)(3,8,7,5) : -28 -> -27] ] ]

5.4 Homomorphisms to a connected groupoid

5.4.1 HomomorphismToSinglePiece

. HomomorphismToSinglePiece(src, rng, piecehoms) (operation)

When G is made up of two or more pieces, all of which get mapped to a connected groupoid, we

have a homomorphism to a single piece. The third input parameter in this case is a list of the individual

homomorphisms from the single pieces (in the correct order!). See section 3.1 for the corresponding

operation on homomorphisms of magmas with objects.

In the following example the source V3 of homV3 has three pieces, and one of the component

homomorphisms is an IdentityMapping .

Example

gap> gend12 := [ (15,16,17,18,19,20), (15,20)(16,19)(17,18) ];;

gap> d12 := Group( gend12 );;

gap> Gd12 := Groupoid( d12, [-37,-36,-35,-34] );;

gap> SetName( d12, "d12" );

gap> SetName( Gd12, "Gd12" );

gap> s3 := Subgroup( d12, [ (15,17,19)(16,18,20), (15,20)(16,19)(17,18) ] );;

gap> Gs3 := SubgroupoidByPieces( Gd12, [ [ s3, [-36,-35,-34] ] ] );;

gap> SetName( s3, "s3" );

gap> SetName( Gs3, "Gs3" );

gap> gend8 := GeneratorsOfGroup( d8 );;

gap> imhd8 := [ ( ), (15,20)(16,19)(17,18) ];;

gap> hd8 := GroupHomomorphismByImages( d8, s3, gend8, imhd8 );;

gap> homd8 := GroupoidHomomorphism( Gd8, Gs3, hd8 );

groupoid homomorphism : Gd8 -> Gs3

[ [ [(1,2,3,4) : -9 -> -9], [(1,3) : -9 -> -9], [() : -9 -> -8],

[() : -9 -> -7] ],

[ [() : -36 -> -36], [(15,20)(16,19)(17,18) : -36 -> -36],

[() : -36 -> -35], [() : -36 -> -34] ] ]

gap> hc6 := GroupHomomorphismByImages( c6, s3,

> [(5,6,7)(8,9)], [(15,16)(17,20)(18,19)] );;

gap> Fs3 := FullSubgroupoid( Gs3, [ -35 ] );;

gap> SetName( Fs3, "Fs3" );

gap> homc6 := GroupoidHomomorphism( Gc6, Fs3, hc6 );;

gap> incFs3 := InclusionMappingGroupoids( Gs3, Fs3 );;

gap> ihomc6 := homc6 * incFs3;

groupoid homomorphism : Gc6 -> Gs3

[ [ [(5,6,7)(8,9) : -6 -> -6] ], [ [(15,16)(17,20)(18,19) : -35 -> -35] ] ]

gap> idGs3 := IdentityMapping( Gs3 );;

gap> V3 := ReplaceOnePieceInUnion( U3, 1, Gs3 );

groupoid with 3 pieces:

[ Gs3, Gd8, Gc6 ]

gap> homs3 := [ idGs3, homd8, ihomc6 ];;

gap> homV3 := HomomorphismToSinglePiece( V3, Gs3, homs3 );;

gap> Display( homV3 );

homomorphism to single piece groupoid with pieces:

(1) : groupoid mapping: [ Gs3 ] -> [ Gs3 ]

root homomorphism: [ [ (15,17,19)(16,18,20), (15,20)(16,19)(17,18) ],

[ (15,17,19)(16,18,20), (15,20)(16,19)(17,18) ] ]

images of objects: [ -36, -35, -34 ]

images of rays: [ [() : -36 -> -36], [() : -36 -> -35], [() : -36 -> -34] ]

(2) : groupoid mapping: [ Gd8 ] -> [ Gs3 ]

root homomorphism: [ [ (1,2,3,4), (1,3) ], [ (), (15,20)(16,19)(17,18) ] ]

images of objects: [ -36, -35, -34 ]

images of rays: [ [() : -36 -> -36], [() : -36 -> -35], [() : -36 -> -34] ]

(3) : groupoid mapping: [ Gc6 ] -> [ Gs3 ]

root homomorphism: [ [ (5,6,7)(8,9) ], [ (15,16)(17,20)(18,19) ] ]

images of objects: [ -35 ]

images of rays: [ [() : -35 -> -35] ]

5.4.2 GroupoidHomomorphismFromHomogeneousDiscrete

. GroupoidHomomorphismFromHomogeneousDiscrete(src, rng, homs, oims) (operation)

This operation requires the source and range; a list of homomorphisms from object group to object

group; and a list of the image objects.

Example

gap> Dd8 := MaximalDiscreteSubgroupoid( Gd8 );

homogeneous, discrete groupoid: < d8, [ -9, -8, -7 ] >

gap> id8 := IdentityMapping( d8 );;

gap> GroupoidHomomorphismFromHomogeneousDiscrete( Dd8, Gd8, [id8,id8,id8],

> [-8,-7,-9] );

groupoid homomorphism : morphism from a homogeneous discrete groupoid:

[ -9, -8, -7 ] -> [ -8, -7, -9 ]

object homomorphisms:

IdentityMapping( d8 )

IdentityMapping( d8 )

IdentityMapping( d8 )

5.5 Homomorphisms to more than one piece

5.5.1 HomomorphismByUnion

. HomomorphismByUnion(src, rng, homs) (operation)

As in section 3.3, when the range H has more than one connected component, a homomorphism

is a union of homomorphisms, one for each piece in the range.

Example

gap> isoq8 := IsomorphismNewObjects( Gq8, [-38,-37] );

groupoid homomorphism :

[ [ [f1 : -28 -> -28], [f2 : -28 -> -28], [f3 : -28 -> -28],

[<identity> of ... : -28 -> -27] ],

[ [f1 : -38 -> -38], [f2 : -38 -> -38], [f3 : -38 -> -38],

[<identity> of ... : -38 -> -37] ] ]

gap> Gq8b := Range( isoq8 );;

gap> SetName( Gq8b, "Gq8b" );

gap> V4 := UnionOfPieces( [ V3, Gq8 ] );

groupoid with 4 pieces:

[ Gs3, Gq8, Gd8, Gc6 ]

gap> SetName( V4, "V4" );

gap> Vs3q8b := UnionOfPieces( [ Gs3, Gq8b ] );;

gap> SetName( Vs3q8b, "Vs3q8b" );

gap> hom4 := HomomorphismByUnion( V4, Vs3q8b, [ homV3, isoq8 ] );;

gap> Display( hom4 );

magma homomorphism: V4 -> Vs3q8b with pieces :

[ Pcgs([ f1, f2, f3 ]) -> [ f1, f2, f3 ], [ -38, -37 ],

[ <identity> of ..., <identity> of ... ] ]

(1) : groupoid mapping: [ Gs3 ] -> [ Gs3 ]

root homomorphism: [ [ (15,17,19)(16,18,20), (15,20)(16,19)(17,18) ],

[ (15,17,19)(16,18,20), (15,20)(16,19)(17,18) ] ]

images of objects: [ -36, -35, -34 ]

images of rays: [ [() : -36 -> -36], [() : -36 -> -35], [() : -36 -> -34] ]

(2) : groupoid mapping: [ Gd8 ] -> [ Gs3 ]

root homomorphism: [ [ (1,2,3,4), (1,3) ], [ (), (15,20)(16,19)(17,18) ] ]

images of objects: [ -36, -35, -34 ]

images of rays: [ [() : -36 -> -36], [() : -36 -> -35], [() : -36 -> -34] ]

(3) : groupoid mapping: [ Gc6 ] -> [ Gs3 ]

root homomorphism: [ [ (5,6,7)(8,9) ], [ (15,16)(17,20)(18,19) ] ]

images of objects: [ -35 ]

images of rays: [ [() : -35 -> -35] ]

5.6 Automorphisms of groupoids

5.6.1 GroupoidAutomorphismByObjectPerm

. GroupoidAutomorphismByObjectPerm(gpd, imobs) (operation)

. GroupoidAutomorphismByGroupAuto(gpd, gpiso) (operation)

. GroupoidAutomorphismByRayShifts(gpd, imrays) (operation)

. GroupoidInnerAutomorphism(gpd, arrow) (operation)

We �rst describe automorphisms of a groupoid G where G is the direct product of a group g and

a complete digraph. The automorphism group is generated by three types of automorphism:

� given a permutation p of the n objects, we de�ne p : G! G; (oi;g;o j) 7! (opi;g;op j);

� given an automorphism a of the root group g, we de�ne a : G! G; (oi;g;o j) 7! (oi;ag;o j);

� given L = [1;g2;g3; : : : ;gn] 2 gn we de�ne qL : G! G; (oi;g;o j) 7! (oi;g
�1
i gg j;o j) so that, in

particular, for all j the rays (r j : o1 ! o j) are shifted by g j : they map to (r jg j : o1 ! o j);

� given g 2G, the inner automorphism of G by g is the mapping h 7! hg as de�ned in section 4.5.

Example

gap> a4 := Subgroup( s4, [(1,2,3),(2,3,4)] );;

gap> SetName( a4, "a4" );

gap> gensa4 := GeneratorsOfGroup( a4 );;

gap> Ga4 := SubgroupoidByPieces( Gs4, [ [a4, [-15,-13,-11]] ] );

single piece groupoid: < a4, [ -15, -13, -11 ] >

gap> SetName( Ga4, "Ga4" );

gap> d := Arrow( Ga4, (1,3,4), -11, -13 );

[(1,3,4) : -11 -> -13]

gap> aut1 := GroupoidAutomorphismByObjectPerm( Ga4, [-13,-11,-15] );;

gap> Display( aut1 );

groupoid mapping: [ Ga4 ] -> [ Ga4 ]

root homomorphism: [ [ (1,2,3), (2,3,4) ], [ (1,2,3), (2,3,4) ] ]

images of objects: [ -13, -11, -15 ]

images of rays: [ [() : -13 -> -13], [() : -13 -> -11], [() : -13 -> -15] ]

gap> d1 := ImageElm( aut1, d );

[(1,3,4) : -15 -> -11]

gap> h2 := GroupHomomorphismByImages( a4, a4, gensa4, [(2,3,4), (1,3,4)] );;

gap> aut2 := GroupoidAutomorphismByGroupAuto( Ga4, h2 );;

gap> Display( aut2 );

groupoid mapping: [ Ga4 ] -> [ Ga4 ]

root homomorphism: [ [ (1,2,3), (2,3,4) ], [ (2,3,4), (1,3,4) ] ]

images of objects: [ -15, -13, -11 ]

images of rays: [ [() : -15 -> -15], [() : -15 -> -13], [() : -15 -> -11] ]

gap> d2 := ImageElm( aut2, d1 );

[(1,2,4) : -15 -> -11]

gap> im3 := [(), (1,3,2), (2,4,3)];;

gap> aut3 := GroupoidAutomorphismByRayShifts( Ga4, im3 );;

gap> Display( aut3 );

groupoid mapping: [ Ga4 ] -> [ Ga4 ]

root homomorphism: [ [ (1,2,3), (2,3,4) ], [ (1,2,3), (2,3,4) ] ]

images of objects: [ -15, -13, -11 ]

images of rays: [ [() : -15 -> -15], [(1,3,2) : -15 -> -13],

[(2,4,3) : -15 -> -11] ]

gap> d3 := ImageElm( aut3, d2 );

[(1,4)(2,3) : -15 -> -11]

gap> d0 := Arrow( Ga4, (2,3,4), -11, -13 );;

gap> aut4 := GroupoidInnerAutomorphism( Ga4, d0 );;

gap> Display( aut4 );

groupoid mapping: [ Ga4 ] -> [ Ga4 ]

root homomorphism: [ [ (1,2,3), (2,3,4) ], [ (1,2,3), (2,3,4) ] ]

images of objects: [ -15, -11, -13 ]

images of rays: [ [() : -15 -> -15], [(2,4,3) : -15 -> -11],

[(2,3,4) : -15 -> -13] ]

gap> d4 := ImageElm( aut4, d3 );

[(1,2,4) : -15 -> -13]

5.6.2 Automorphisms of a groupoid with rays

If S is a wide subgroupoid with rays of a standard groupoid G then an automorphism a of Gmay map

S to a different subgroupoid. When a is a GroupoidAutomorphismByObjectPerm, then it restricts

to an automorphism of S provided the objects in each orbit have the same ray elements.

This subsection needs to be expanded and examples added.

5.6.3 AutomorphismGroupOfGroupoid

. AutomorphismGroupOfGroupoid(gpd) (operation)

. NiceObjectAutoGroupGroupoid(gpd, aut) (operation)

The AutomorphismGroup of G is isomorphic to the quotient of Sn�A� gn by a subgroup iso-

morphic to g, where A is the automorphism group of g and Sn is the symmetric group on the n objects.

This is one of the main topics in [AW10].

The function NiceObjectAutoGroupGroupoid takes a groupoid and a subgroup of its automor-

phism group and retuns a nice monomorphism from this automorphism group to a pc-group, if one is

available. The current implementation is experimental. Note that ImageElm at present only works on

generating elements.

Example

gap> AGa4 := AutomorphismGroupOfGroupoid( Ga4 );

<group with 8 generators>

gap> AGgens := GeneratorsOfGroup( AGa4);;

gap> NGa4 := NiceObject( AGa4 );;

gap> MGa4 := NiceMonomorphism( AGa4 );;

gap> Size( AGa4 );

20736

gap> SetName( AGa4, "AGa4" );

gap> SetName( NGa4, "NGa4" );

gap> ## cannot test images of AGgens because of random variations

gap> ## Now do some tests!

gap> mgi := MappingGeneratorsImages( MGa4 );;

gap> autgen := mgi[1];;

gap> pcgen := mgi[2];;

gap> ngen := Length( autgen );;

gap> ForAll( [1..ngen], i -> Order(autgen[i]) = Order(pcgen[i]) );

true

5.6.4 Inner automorphisms

The inner automorphism subgroup Inn(G) of the automorphism group of G is the group of inner

automorphisms ^a : b 7! ba for a 2 G. It is not the case that the map G! Inn(G);a 7! ^a preserves

multiplication. Indeed, when a= (o;g; p);b= (p;h;r) 2 G with objects p;q;r all distict, then

^(ab) = (^a)(^b)(^a) = (^b)(^a)(^b):

(Compare this with the permutation identity (pq)(qr)(pq) = (pr) = (qr)(pq)(qr).) So the map G!

Inn(G) is of type IsMappingWithObjectsByFunction.
In the example we convert the automorphism group AGa4 into a single object groupoid, and then

de�ne the inner automorphism map.

Example

gap> AGa40 := Groupoid( AGa4, [0] );

single piece groupoid: < AGa4, [ 0 ] >

gap> conj := function(a)

> return ArrowNC( true, GroupoidInnerAutomorphism(Ga4,a), 0, 0 );

> end;;

gap> inner := MappingWithObjectsByFunction( Ga4, AGa40, conj, [0,0,0] );;

gap> a1 := Arrow( Ga4, (1,2,3), -15, -13 );;

gap> inn1 := ImageElm( inner, a1 );;

gap> a2 := Arrow( Ga4, (2,3,4), -13, -11 );;

gap> inn2 := ImageElm( inner, a2 );;

gap> a3 := a1*a2;

[(1,3)(2,4) : -15 -> -11]

gap> inn3 := ImageElm( inner, a3 );

[groupoid homomorphism : Ga4 -> Ga4

[ [ [(1,2,3) : -15 -> -15], [(2,3,4) : -15 -> -15], [() : -15 -> -13],

[() : -15 -> -11] ],

[ [(1,3,4) : -11 -> -11], [(1,2,4) : -11 -> -11], [(1,3)(2,4) : -11 -> -13],

[() : -11 -> -15] ] ] : 0 -> 0]

gap> (inn3 = inn1*inn2*inn1) and (inn3 = inn2*inn1*inn2);

true

5.6.5 GroupoidAutomorphismByGroupAutos

. GroupoidAutomorphismByGroupAutos(gpd, auts) (operation)

Homogeneous, discrete groupoids are the second type of groupoid for which a method

is provided for AutomorphismGroupOfGroupoid. This is used in the XMod package for

constructing crossed modules of groupoids. The two types of generating automorphism are

GroupoidAutomorphismByGroupAutos, which requires a list of group automorphisms, one for each

object group, and GroupoidAutomorphismByObjectPerm, which permutes the objects.

Example

gap> Hs3 := HomogeneousDiscreteGroupoid( s3, [ -13..-10] );

homogeneous, discrete groupoid: < s3, [ -13 .. -10 ] >

gap> aut4 := GroupoidAutomorphismByObjectPerm( Hs3, [-12,-10,-11,-13] );

groupoid homomorphism : morphism from a homogeneous discrete groupoid:

[ -13, -12, -11, -10 ] -> [ -12, -10, -11, -13 ]

object homomorphisms:

IdentityMapping( s3 )

IdentityMapping( s3 )

IdentityMapping( s3 )

IdentityMapping( s3 )

gap> gens3 := GeneratorsOfGroup( s3 );;

gap> g1 := gens3[1];;

gap> g2 := gens3[2];;

gap> b1 := GroupHomomorphismByImages( s3, s3, gens3, [g1, g2^g1 ] );;

gap> b2 := GroupHomomorphismByImages( s3, s3, gens3, [g1^g2, g2 ] );;

gap> b3 := GroupHomomorphismByImages( s3, s3, gens3, [g1^g2, g2^(g1*g2) ] );;

gap> b4 := GroupHomomorphismByImages( s3, s3, gens3, [g1^(g2*g1), g2^g1 ] );;

gap> aut5 := GroupoidAutomorphismByGroupAutos( Hs3, [b1,b2,b3,b4] );

groupoid homomorphism : morphism from a homogeneous discrete groupoid:

[ -13, -12, -11, -10 ] -> [ -13, -12, -11, -10 ]

object homomorphisms:

GroupHomomorphismByImages( s3, s3,

[ (15,17,19)(16,18,20), (15,20)(16,19)(17,18) ],

[ (15,17,19)(16,18,20), (15,18)(16,17)(19,20) ] )

GroupHomomorphismByImages( s3, s3,

[ (15,17,19)(16,18,20), (15,20)(16,19)(17,18) ],

[ (15,19,17)(16,20,18), (15,20)(16,19)(17,18) ] )

GroupHomomorphismByImages( s3, s3,

[ (15,17,19)(16,18,20), (15,20)(16,19)(17,18) ],

[ (15,19,17)(16,20,18), (15,16)(17,20)(18,19) ] )

GroupHomomorphismByImages( s3, s3,

[ (15,17,19)(16,18,20), (15,20)(16,19)(17,18) ],

[ (15,19,17)(16,20,18), (15,18)(16,17)(19,20) ] )

gap> AHs3 := AutomorphismGroupOfGroupoid( Hs3 );

<group with 4 generators>

gap> Size( AHs3 );

31104

gap> genAHs3 := GeneratorsOfGroup( AHs3 );;

gap> Length( genAHs3 );

4

gap> nobAHs3 := NiceObject( AHs3 );;

gap> nmonAHs3 := NiceMonomorphism( AHs3 );;

gap> w := genAHs3[1];;

gap> w1 := ImageElm( nmonAHs3, w );;

gap> x := genAHs3[2];;

gap> x1 := ImageElm( nmonAHs3, x );;

gap> y := genAHs3[3];;

gap> y1 := ImageElm( nmonAHs3, y );;

gap> z := genAHs3[4];;

gap> z1 := ImageElm( nmonAHs3, z );;

gap> u := z*w*y*x*z;

groupoid homomorphism : morphism from a homogeneous discrete groupoid:

[ -13, -12, -11, -10 ] -> [ -11, -13, -10, -12 ]

object homomorphisms:

IdentityMapping( s3 )

ConjugatorAutomorphism( s3, (15,19,17)(16,20,18) )

IdentityMapping( s3 )

ConjugatorAutomorphism( s3, (15,20)(16,19)(17,18) )

gap> u1 := z1*w1*y1*x1*z1;

(1,2,4,3)(5,17,23,16,8,20,26,13)(6,18,24,15,7,19,25,14)(9,21,27,12,10,22,28,

11)

gap> imu := ImageElm( nmonAHs3, u );;

gap> u1 = imu;

true

5.7 Matrix representations of groupoids

Suppose that gpd is the direct product of a group G and a complete digraph, and that r : G!M is an

isomorphism to a matrix group M. Then if rep is the isomorphic groupoid with the same objects and

root group M there is an isomorphism m from gpd to rep mapping (g : i! j) to (rg : i! j).
When gpd is a groupoid with rays, a representation can be obtained by restricting a representation

of its parent.

Example

gap> reps := IrreducibleRepresentations( s4 );;

gap> rep4 := reps[4];;

gap> Rs4 := Groupoid( Image( rep4 ), ObjectList( Gs4 ) );

single piece groupoid: < Group([ [ [ 0, 1, 0 ], [ 1, 0, 0 ], [ 0, 0, 1 ] ],

[ [ 0, 0, 1 ], [ 1, 0, 0 ], [ 0, 1, 0 ] ],

[ [ -1, 0, 0 ], [ 0, 1, 0 ], [ 0, 0, -1 ] ],

[ [ 1, 0, 0 ], [ 0, -1, 0 ], [ 0, 0, -1 ] ] ]), [ -15, -14, -13, -12, -11

] >

gap> IsMatrixGroupoid( Rs4 );

true

gap> gens := GeneratorsOfGroupoid( Gs4 );

[ [(1,2,3,4) : -15 -> -15], [(3,4) : -15 -> -15], [() : -15 -> -14],

[() : -15 -> -13], [() : -15 -> -12], [() : -15 -> -11] ]

gap> images := List( gens,

> g -> Arrow( Rs4, ImageElm(rep4,g![1]), g![2], g![3] ) );

[ [[ [ -1, 0, 0 ], [ 0, 0, 1 ], [ 0, -1, 0 ] ] : -15 -> -15],

[[ [ 0, 1, 0 ], [ 1, 0, 0 ], [ 0, 0, 1 ] ] : -15 -> -15],

[[ [ 1, 0, 0 ], [ 0, 1, 0 ], [ 0, 0, 1 ] ] : -15 -> -14],

[[ [ 1, 0, 0 ], [ 0, 1, 0 ], [ 0, 0, 1 ] ] : -15 -> -13],

[[ [ 1, 0, 0 ], [ 0, 1, 0 ], [ 0, 0, 1 ] ] : -15 -> -12],

[[ [ 1, 0, 0 ], [ 0, 1, 0 ], [ 0, 0, 1 ] ] : -15 -> -11] ]

gap> mor := GroupoidHomomorphismFromSinglePiece( Gs4, Rs4, gens, images );

groupoid homomorphism :

[ [ [(1,2,3,4) : -15 -> -15], [(3,4) : -15 -> -15], [() : -15 -> -14],

[() : -15 -> -13], [() : -15 -> -12], [() : -15 -> -11] ],

[ [[ [ -1, 0, 0 ], [ 0, 0, 1 ], [ 0, -1, 0 ] ] : -15 -> -15],

[[ [ 0, 1, 0 ], [ 1, 0, 0 ], [ 0, 0, 1 ] ] : -15 -> -15],

[[ [ 1, 0, 0 ], [ 0, 1, 0 ], [ 0, 0, 1 ] ] : -15 -> -14],

[[ [ 1, 0, 0 ], [ 0, 1, 0 ], [ 0, 0, 1 ] ] : -15 -> -13],

[[ [ 1, 0, 0 ], [ 0, 1, 0 ], [ 0, 0, 1 ] ] : -15 -> -12],

[[ [ 1, 0, 0 ], [ 0, 1, 0 ], [ 0, 0, 1 ] ] : -15 -> -11] ] ]

gap> a := Arrow( Hs4, (1,4,2), -12, -13 );

[(1,4,2) : -12 -> -13]

gap> ImageElm( mor, a );

[[ [ 0, 0, -1 ], [ -1, 0, 0 ], [ 0, 1, 0 ] ] : -12 -> -13]

gap> rmor := RestrictedMappingGroupoids( mor, Hd8b );

groupoid homomorphism :

[ [ [(1,2,3,4) : -14 -> -14], [(1,3) : -14 -> -14], [(1,2,3) : -14 -> -13],

[(1,2,4) : -14 -> -12] ],

[ [[ [ -1, 0, 0 ], [ 0, 0, 1 ], [ 0, -1, 0 ] ] : -14 -> -14],

[[ [ 1, 0, 0 ], [ 0, 0, -1 ], [ 0, -1, 0 ] ] : -14 -> -14],

[[ [ 0, 0, 1 ], [ -1, 0, 0 ], [ 0, -1, 0 ] ] : -14 -> -13],

[[ [ 0, -1, 0 ], [ 0, 0, 1 ], [ -1, 0, 0 ] ] : -14 -> -12] ] ]



Chapter 6

Graphs of Groups and Groupoids

This package was originally designed to implement graphs of groups, a notion introduced by Serre in

[Ser80]. It was only when this was extended to graphs of groupoids that the functions for groupoids,

described in the previous chapters, were required. The methods described here are based on Philip

Higgins' paper [Hig76]. For further details see Chapter 2 of [Moo01]. Since a graph of groups

involves a directed graph, with a group associated to each vertex and arc, we �rst de�ne digraphs with

edges weighted by the generators of a free group.

6.1 Digraphs

6.1.1 FpWeightedDigraph

. FpWeightedDigraph(verts, arcs) (attribute)

. IsFpWeightedDigraph(dig) (attribute)

. InvolutoryArcs(dig) (attribute)

A weighted digraph is a record with two components: vertices, which are usually taken to be

positive integers (to distinguish them from the objects in a groupoid); and arcs, which take the form

of 3-element lists [weight,tail,head]. The tail and head are the two vertices of the arc. The

weight is taken to be an element of a �nitely presented group, so as to produce digraphs of type

IsFpWeightedDigraph.

Example

gap> V1 := [ 5, 6 ];;

gap> fg1 := FreeGroup( "y" );;

gap> y := fg1.1;;

gap> A1 := [ [ y, 5, 6 ], [ y^-1, 6, 5 ] ];

gap> D1 := FpWeightedDigraph( fg1, V1, A1 );

weighted digraph with vertices: [ 5, 6 ]

and arcs: [ [ y, 5, 6 ], [ y^-1, 6, 5 ] ]

gap> inv1 := InvolutoryArcs( D1 );

[ 2, 1 ]

The example illustrates the fact that we require arcs to be de�ned in involutory pairs, as though they

were inverse elements in a groupoid. We may in future decide just to give [y,5,6] as the data and

46

get the function to construct the reverse edge. The attribute InvolutoryArcs returns a list of the

positions of each inverse arc in the list of arcs. In the second example the graph is a complete digraph

on three vertices.
Example

gap> fg3 := FreeGroup( 3, "z" );;

gap> z1 := fg3.1;; z2 := fg3.2;; z3 := fg3.3;;

gap> ob3 := [ 7, 8, 9 ];;

gap> A3 := [[z1,7,8],[z2,8,9],[z3,9,7],[z1^-1,8,7],[z2^-1,9,8],[z3^-1,7,9]];;

gap> D3 := FpWeightedDigraph( fg3, ob3, A3 );

weighted digraph with vertices: [ 7, 8, 9 ]

and arcs: [ [ z1, 7, 8 ], [ z2, 8, 9 ], [ z3, 9, 7 ], [ z1^-1, 8, 7 ],

[ z2^-1, 9, 8 ], [ z3^-1, 7, 9 ] ]

[gap> inob3 := InvolutoryArcs( D3 );

[ 4, 5, 6, 1, 2, 3 ]

6.2 Graphs of Groups

6.2.1 GraphOfGroups

. GraphOfGroups(dig, gps, isos) (operation)

. DigraphOfGraphOfGroups(gg) (attribute)

. GroupsOfGraphOfGroups(gg) (attribute)

. IsomorphismsOfGraphOfGroups(gg) (attribute)

A graph of groups is traditionally de�ned as consisting of:

� a digraph with involutory pairs of arcs;

� a vertex group associated to each vertex;

� a group associated to each pair of arcs;

� an injective homomorphism from each arc group to the group at the head of the arc.

We have found it more convenient to associate to each arc:

� a subgroup of the vertex group at the tail;

� a subgroup of the vertex group at the head;

� an isomorphism between these subgroups, such that each involutory pair of arcs determines

inverse isomorphisms.

These two viewpoints are clearly equivalent.

In this implementation we require that all subgroups are of �nite index in the vertex groups.

The three attributes provide a means of calling the three items of data in the construction of a

graph of groups.

We shall be representing free products with amalgamation of groups and HNN extensions of

groups, so we take as our �rst example the trefoil group with generators a;b and relation a3 = b2.

For this we take digraph D1 above with an in�nite cyclic group at each vertex, generated by a and b

respectively. The two subgroups will be generated by a3 and b2 with the obvious isomorphisms.

Example

gap> ## free vertex group at 5

gap> fa := FreeGroup( "a" );;

gap> a := fa.1;;

gap> SetName( fa, "fa" );

gap> hy := Subgroup( fa, [a^3] );;

gap> SetName( hy, "hy" );

gap> ## free vertex group at 6

gap> fb := FreeGroup( "b" );;

gap> b := fb.1;;

gap> SetName( fb, "fb" );

gap> hybar := Subgroup( fb, [b^2] );;

gap> SetName( hybar, "hybar" );

gap> ## isomorphisms between subgroups

gap> homy := GroupHomomorphismByImagesNC( hy, hybar, [a^3], [b^2] );;

gap> homybar := GroupHomomorphismByImagesNC( hybar, hy, [b^2], [a^3] );;

gap> ## defining graph of groups G1

gap> G1 := GraphOfGroups( D1, [fa,fb], [homy,homybar] );

Graph of Groups: 2 vertices; 2 arcs; groups [ fa, fb ]

gap> Display( G1 );

Graph of Groups with :-

vertices: [ 5, 6 ]

arcs: [ [ y, 5, 6 ], [ y^-1, 6, 5 ] ]

groups: [ fa, fb ]

isomorphisms: [ [ [ a^3 ], [ b^2 ] ], [ [ b^2 ], [ a^3 ] ] ]

6.2.2 IsGraphOfFpGroups

. IsGraphOfFpGroups(gg) (property)

. IsGraphOfPcGroups(gg) (property)

. IsGraphOfPermGroups(gg) (property)

This is a list of properties to be expected of a graph of groups. In principle any type of group

known to GAP may be used as vertex groups, though these types are not normally mixed in a single

structure.
Example

gap> IsGraphOfFpGroups( G1 );

true

gap> IsomorphismsOfGraphOfGroups( G1 );

[ [ a^3 ] -> [ b^2 ], [ b^2 ] -> [ a^3 ] ]

6.2.3 RightTransversalsOfGraphOfGroups

. RightTransversalsOfGraphOfGroups(gg) (attribute)

. LeftTransversalsOfGraphOfGroups(gg) (attribute)

Computation with graph of groups words will require, for each arc subgroup ha, a set of represen-

tatives for the left cosets of ha in the tail vertex group. As already pointed out, we require subgroups

of �nite index. Since GAP prefers to provide right cosets, we obtain the right representatives �rst,

and then invert them.

When the vertex groups are of type FpGroup we shall require normal forms for these groups, so

we assume that such vertex groups are provided with Knuth Bendix rewriting systems using functions

from the main GAP library, (e.g. IsomorphismFpSemigroup).

Example

gap> RTG1 := RightTransversalsOfGraphOfGroups( G1 );

[ [ <identity ...>, a, a^2 ], [ <identity ...>, b ] ]

gap> LTG1 := LeftTransversalsOfGraphOfGroups( G1 );

[ [ <identity ...>, a^-1, a^-2 ], [ <identity ...>, b^-1 ] ]

6.3 Words in a Graph of Groups and their normal forms

6.3.1 GraphOfGroupsWord

. GraphOfGroupsWord(gg, tv, list) (operation)

. IsGraphOfGroupsWord(w) (property)

. GraphOfGroupsOfWord(w) (attribute)

. WordOfGraphOfGroupsWord(w) (attribute)

. TailOfGraphOfGroupsWord(w) (attribute)

. HeadOfGraphOfGroupsWord(w) (attribute)

If G is a graph of groups with underlying digraph D, the following groupoids may be considered.

First there is the free groupoid or path groupoid on D. Since we want each involutory pair of arcs to

represent inverse elements in the groupoid, we quotient out by the relations y^-1 = ybar to obtain

PG(D). Secondly, there is the discrete groupoid VG(D), namely the union of all the vertex groups.

Since these two groupoids have the same object set (the vertices of D) we can form A(G), the free

product of PG(D) and VG(D) amalgamated over the vertices. For further details of this universal

groupoid construction see [Moo01]. (Note that these groupoids are not implemented in this package.)

An element of A(G) is a graph of groups word which may be represented by a list of the form

w = [g1;y1;g2;y2; :::;gn;yn;gn+1]. Here each yi is an arc of D; the head of yi�1 is a vertex vi which is

also the tail of yi; and gi is an element of the vertex group at vi.

So a graph of groups word requires as data the graph of groups; the tail vertex for the word; and a

list of arcs and group elements. We may specify each arc by its position in the list of arcs.

In the following example, where gw1 is a word in the trefoil graph of groups, the yi are speci�ed

by their positions in A1. Both arcs are traversed twice, so the resulting word is a loop at vertex 5.

Example

gap> L1 := [ a^7, 1, b^-6, 2, a^-11, 1, b^9, 2, a^7 ];;

gap> gw1 := GraphOfGroupsWord( G1, 5, L1 );

(5)a^7.y.b^-6.y^-1.a^-11.y.b^9.y^-1.a^7(5)

gap> IsGraphOfGroupsWord( gw1 );

true

gap> [ TailOfGraphOfGroupsWord(gw1), HeadOfGraphOfGroupsWord(gw1) ];

[ 5, 5 ]

gap> GraphOfGroupsOfWord(gw1);

Graph of Groups: 2 vertices; 2 arcs; groups [ fa, fb ]

gap> WordOfGraphOfGroupsWord( gw1 );

[ a^7, 1, b^-6, 2, a^-11, 1, b^9, 2, a^7 ]

6.3.2 ReducedGraphOfGroupsWord

. ReducedGraphOfGroupsWord(w) (operation)

. IsReducedGraphOfGroupsWord(w) (property)

A graph of groups word may be reduced in two ways, to give a normal form. Firstly, if part of

the word has the form [yi, identity, yibar] then this subword may be omitted. This is known

as a length reduction. Secondly there are coset reductions. Working from the left-hand end of the

word, subwords of the form [gi;yi;gi+1] are replaced by [ti;yi;mi(hi) � gi+1] where gi = ti � hi is the

unique factorisation of gi as a left coset representative times an element of the arc subgroup, and mi

is the isomorphism associated to yi. Thus we may consider a coset reduction as passing a subgroup

element along an arc. The resulting normal form (if no length reductions have taken place) is then

[t1;y1; t2;y2; :::; tn;yn;k] for some k in the head group of yn. For further details see Section 2.2 of

[Moo01].

The reduction of the word gw1 in our example includes one length reduction. The four stages of

the reduction are as follows:

a7b�6a�11b9a7 7! a�2b0a�11b9a7 7! a�13b9a7 7! a�1b�8b9a7 7! a�1b�1a10:

Example

gap> nw1 := ReducedGraphOfGroupsWord( gw1 );

(5)a^-1.y.b^-1.y^-1.a^10(5)

6.4 Free products with amalgamation and HNN extensions

6.4.1 FreeProductWithAmalgamation

. FreeProductWithAmalgamation(gp1, gp2, iso) (operation)

. IsFpaGroup(fpa) (property)

. GraphOfGroupsRewritingSystem(fpa) (attribute)

. NormalFormGGRWS(fpa, word) (attribute)

As we have seen with the trefoil group example, graphs of groups can be used to obtain a normal

form for free products with amalgamation G1 �H G2 when G1;G2 both have rewrite systems, and H is

of �nite index in both G1 and G2.

When gp1 and gp2 are fp-groups, the operation FreeProductWithAmalgamation constructs

the required fp-group. When the two groups are permutation groups, the IsomorphismFpGroup

operation is called on both gp1 and gp2, and the resulting isomorphism is transported to one between

the two new subgroups.

The attribute GraphOfGroupsRewritingSystem of fpa is the graph of groups which has under-

lying digraph D1, with two vertices and two arcs; the two groups as vertex groups; and the speci�ed

isomorphisms on the arcs. Despite the name, graphs of groups constructed in this way do not belong

to the category IsRewritingSystem. This anomaly may be dealt with when time permits.

The example below shows a computation in the the free product of the symmetric s3 and the

alternating a4, amalgamated over a cyclic subgroup c3.

Example

gap> ## set up the first group s3 and a subgroup c3=<a1>

gap> fg2 := FreeGroup( 2, "a" );;

gap> rel1 := [ fg2.1^3, fg2.2^2, (fg2.1*fg2.2)^2 ];;

gap> s3 := fg2/rel1;;

gap> gs3 := GeneratorsOfGroup(s3);;

gap> SetName( s3, "s3" );

gap> a1 := gs3[1];; a2 := gs3[2];;

gap> H1 := Subgroup(s3,[a1]);;

gap> ## then the second group a4 and subgroup c3=<b1>

gap> f2 := FreeGroup( 2, "b" );;

gap> rel2 := [ f2.1^3, f2.2^3, (f2.1*f2.2)^2 ];;

gap> a4 := f2/rel2;;

gap> ga4 := GeneratorsOfGroup(a4);;

gap> SetName( a4, "a4" );

gap> b1 := ga4[1]; b2 := ga4[2];;

gap> H2 := Subgroup(a4,[b1]);;

gap> ## form the isomorphism and the fpa group

gap> iso := GroupHomomorphismByImages(H1,H2,[a1],[b1]);;

gap> fpa := FreeProductWithAmalgamation( s3, a4, iso );

<fp group on the generators [ fa1, fa2, fa3, fa4 ]>

gap> RelatorsOfFpGroup( fpa );

[ fa1^3, fa2^2, (fa1*fa2)^2, fa3^3, fa4^3, (fa3*fa4)^2, fa1*fa3^-1 ]

gap> gg1 := GraphOfGroupsRewritingSystem( fpa );;

gap> Display( gg1 );

Graph of Groups with :-

vertices: [ 5, 6 ]

arcs: [ [ y, 5, 6 ], [ y^-1, 6, 5 ] ]

groups: [ s3, a4 ]

isomorphisms: [ [ [ a1 ], [ b1 ] ], [ [ b1 ], [ a1 ] ] ]

gap> LeftTransversalsOfGraphOfGroups( gg1 );

[ [ <identity ..>, a2^-1 ], [ <identity ..>, b2^-1, b1^-1*b2^-1, b1*b2^-1 ] ]

gap> ## choose a word in fpa and find its normal form

gap> gfpa := GeneratorsOfGroup( fpa );;

gap> w2 := (gfpa[1]*gfpa[2]*gfpa[3]^gfpa[4])^3;

(fa1*fa2*fa4^-1*fa3*fa4)^3

gap> n2 := NormalFormGGRWS( fpa, w2 );

fa2*fa3*(fa4^-1*fa2)^2*fa4^-1*fa3

6.4.2 HnnExtension

. HnnExtension(gp, iso) (operation)

. IsHnnGroup(hnn) (property)

For HNN extensions, the appropriate graph of groups has underlying digraph with just one vertex

and one pair of loops, weighted with FpGroup generators z;z�1. There is one vertex group G, two

isomorphic subgroups H1,H2 of G, with the isomorphism and its inverse on the loops. The presentation

of the extension has one more generator than that of G and corresponds to the generator z.

The functions GraphOfGroupsRewritingSystem and NormalFormGGRWSmay be applied to hnn-

groups as well as to fpa-groups.

In the example we take G=a4 and the two subgroups are cyclic groups of order 3.

Example

gap> H3 := Subgroup(a4,[b2]);;

gap> i23 := GroupHomomorphismByImages( H2, H3, [b1], [b2] );;

gap> hnn := HnnExtension( a4, i23 );

<fp group on the generators [ fe1, fe2, fe3 ]>

gap> phnn := PresentationFpGroup( hnn );;

gap> TzPrint( phnn );

#I generators: [ fe1, fe2, fe3 ]

#I relators:

#I 1. 3 [ 1, 1, 1 ]

#I 2. 3 [ 2, 2, 2 ]

#I 3. 4 [ 1, 2, 1, 2 ]

#I 4. 4 [ -3, 1, 3, -2 ]

gap> gg2 := GraphOfGroupsRewritingSystem( hnn );

Graph of Groups: 1 vertices; 2 arcs; groups [ a4 ]

gap> LeftTransversalsOfGraphOfGroups( gg2 );

[ [ <identity ...>, b2^-1, b1^-1*b2^-1, b1*b2^-1 ],

[ <identity ...>, b1^-1, b1, b2^-1*b1 ] ]

gap> gh := GeneratorsOfGroup( hnn );;

gap> w3 := (gh[1]^gh[2])*gh[3]^-1*(gh[1]*gh[3]*gh[2]^2)^2*gh[3]*gh[2];

fe2^-1*fe1*fe2*fe3^-1*(fe1*fe3*fe2^2)^2*fe3*fe2

gap> n3 := NormalFormGGRWS( hnn, w3 );

(fe2*fe1*fe3)^2

Both fpa-groups and hnn-groups are provided with a record attribute, FpaInfo(fpa) and

HnnInfo(hnn) respectively, storing the groups and isomorphisms involved in their construction.

Example

gap> fpainfo := FpaInfo( fpa );

rec( groups := [ s3, a4 ], positions := [ [ 1, 2 ], [ 3, 4 ] ],

isomorphism := [ a1 ] -> [ b1 ] )

gap> hnninfo := HnnInfo( hnn );

rec( group := a4, isomorphism := [ b1 ] -> [ b2 ] )

6.5 GraphsOfGroupoids and their Words

6.5.1 GraphOfGroupoids

. GraphOfGroupoids(dig, gpds, subgpds, isos) (operation)

. IsGraphOfPermGroupoids(gg) (property)

. IsGraphOfFpGroupoids(gg) (property)

. GroupoidsOfGraphOfGroupoids(gg) (attribute)

. DigraphOfGraphOfGroupoids(gg) (attribute)

. SubgroupoidsOfGraphOfGroupoids(gg) (attribute)

. IsomorphismsOfGraphOfGroupoids(gg) (attribute)

. RightTransversalsOfGraphOfGroupoids(gg) (attribute)

. LeftTransversalsOfGraphOfGroupoids(gg) (attribute)

Graphs of groups generalise naturally to graphs of groupoids, forming the class

IsGraphOfGroupoids. There is now a groupoid at each vertex and the isomorphism on an arc iden-

ti�es wide subgroupoids at the tail and at the head. Since all subgroupoids are wide, every groupoid

in a connected constituent of the graph has the same number of objects, but there is no requirement

that the object sets are all the same.

The example below generalises the trefoil group example in subsection 4.4.1, taking at each vertex

of D1 a two-object groupoid with a free group on one generator, and full subgroupoids with groups

ha3i and hb2i.
Example

gap> Gfa := SinglePieceGroupoid( fa, [-2,-1] );;

gap> ofa := One( fa );;

gap> SetName( Gfa, "Gfa" );

gap> Uhy := Subgroupoid( Gfa, [ [ hy, [-2,-1] ] ] );;

gap> SetName( Uhy, "Uhy" );

gap> Gfb := SinglePieceGroupoid( fb, [-4,-3] );;

gap> ofb := One( fb );;

gap> SetName( Gfb, "Gfb" );

gap> Uhybar := Subgroupoid( Gfb, [ [ hybar, [-4,-3] ] ] );;

gap> SetName( Uhybar, "Uhybar" );

gap> gens := GeneratorsOfGroupoid( Uhy );;

gap> gensbar := GeneratorsOfGroupoid( Uhybar );;

gap> mory := GroupoidHomomorphismFromSinglePiece(

> Uhy, Uhybar, gens, gensbar );

groupoid homomorphism : Uhy -> Uhybar

[ [ [a^3 : -2 -> -2], [<identity ...> : -2 -> -1] ],

[ [b^2 : -4 -> -4], [<identity ...> : -4 -> -3] ] ]

gap> morybar := InverseGeneralMapping( mory );

groupoid homomorphism : Uhybar -> Uhy

[ [ [b^2 : -4 -> -4], [<identity ...> : -4 -> -3] ],

[ [a^3 : -2 -> -2], [<identity ...> : -2 -> -1] ] ]

gap> gg3 := GraphOfGroupoids( D1, [Gfa,Gfb], [Uhy,Uhybar], [mory,morybar] );;

gap> Display( gg3 );

Graph of Groupoids with :-

vertices: [ 5, 6 ]

arcs: [ [ y, 5, 6 ], [ y^-1, 6, 5 ] ]

groupoids:

fp single piece groupoid: Gfa

objects: [ -2, -1 ]

group: fa = <[ a ]>

fp single piece groupoid: Gfb

objects: [ -4, -3 ]

group: fb = <[ b ]>

subgroupoids: single piece groupoid: Uhy

objects: [ -2, -1 ]

group: hy = <[ a^3 ]>

single piece groupoid: Uhybar

objects: [ -4, -3 ]

group: hybar = <[ b^2 ]>

isomorphisms: [ groupoid homomorphism : Uhy -> Uhybar

[ [ [a^3 : -2 -> -2], [<identity ...> : -2 -> -1] ],

[ [b^2 : -4 -> -4], [<identity ...> : -4 -> -3] ] ],

groupoid homomorphism : Uhybar -> Uhy

[ [ [b^2 : -4 -> -4], [<identity ...> : -4 -> -3] ],

[ [a^3 : -2 -> -2], [<identity ...> : -2 -> -1] ] ] ]

6.5.2 GraphOfGroupoidsWord

. GraphOfGroupoidsWord(gg, tv, list) (operation)

. IsGraphOfGroupoidsWord(w) (property)

. GraphOfGroupoidsOfWord(w) (attribute)

. WordOfGraphOfGroupoidsWord(w) (attribute)

. ReducedGraphOfGroupoidsWord(w) (operation)

. IsReducedGraphOfGroupoidsWord(w) (property)

Having produced the graph of groupoids gg3, we may construct left coset representatives; choose

a graph of groupoids word; and reduce this to normal form. Compare the nw3 below with the normal

form nw1 in subsection 4.3.2.
Example

gap> f1 := Arrow( Gfa, a^7, -1, -2);;

gap> f2 := Arrow( Gfb, b^-6, -4, -4 );;

gap> f3 := Arrow( Gfa, a^-11, -2, -1 );;

gap> f4 := Arrow( Gfb, b^9, -3, -4 );;

gap> f5 := Arrow( Gfa, a^7, -2, -1 );;

gap> L3 := [ f1, 1, f2, 2, f3, 1, f4, 2, f5 ];

[ [a^7 : -1 -> -2], 1, [b^-6 : -4 -> -4], 2, [a^-11 : -2 -> -1], 1,

[b^9 : -3 -> -4], 2, [a^7 : -2 -> -1] ]

gap> gw3 := GraphOfGroupoidsWord( gg3, 5, L3);

(5)[a^7 : -1 -> -2].y.[b^-6 : -4 -> -4].y^-1.[a^-11 : -2 -> -1].y.[b^9 :

-3 -> -4].y^-1.[a^7 : -2 -> -1](5)

gap> nw3 := ReducedGraphOfGroupoidsWord( gw3 );

(5)[a^-1 : -1 -> -2].y.[b^-1 : -4 -> -4].y^-1.[a^10 : -2 -> -1](5)

More examples of these operations may be found in the example �le

groupoids/examples/ggraph.g.



Chapter 7

Technical Notes

This short chapter is included for the bene�t of anyone wishing to implement some other variety

of many-object structures, for example ringoids, which are rings with many objects; Lie groupoids,

which are Lie groups with many objects; and so on.

7.1 Many object structures

Structures with many objects, and their elements, are de�ned in a manner similar to the single object

case. For elements we have:

� DeclareCategory( "IsMultiplicativeElementWithObjects",

IsMultiplicativeElement );

� DeclareCategory( "IsMultiplicativeElementWithObjectsAndOnes",

IsMultiplicativeElementWithObjects );

� DeclareCategory( "IsMultiplicativeElementWithObjectsAndInverses",

IsMultiplicativeElementWithObjectsAndOnes );

� DeclareCategory( "IsGroupoidElement",

IsMultiplicativeElementWithObjectsAndInverses );

as well as various category collections. For the various structures we have:

� DeclareCategory( "IsDomainWithObjects", IsDomain );

� DeclareCategory( "IsMagmaWithObjects", IsDomainWithObjects and

IsMultiplicativeElementWithObjectsCollection );

� DeclareCategory( "IsMagmaWithObjectsAndOnes", IsMagmaWithObjects and

IsMultiplicativeElementWithObjectsAndOnesCollection );

� DeclareCategory( "IsMagmaWithObjectsAndInverses",

IsMagmaWithObjectsAndOnes and

IsMultiplicativeElementWithObjectsAndInversesCollection );

� DeclareCategory( "IsGroupoid", IsMagmaWithObjectsAndInverses and

IsGroupoidElementCollection );

55

Among the groupoids constructed earlier are the single piece Gd8 and the �ve component union U5:

Example

gap> CategoriesOfObject( Gd8 );

[ "IsListOrCollection", "IsCollection", "IsExtLElement",

"CategoryCollections(IsExtLElement)", "IsExtRElement",

"CategoryCollections(IsExtRElement)",

"CategoryCollections(IsMultiplicativeElement)", "IsGeneralizedDomain",

"IsMagma", "IsDomainWithObjects",

"CategoryCollections(IsMultiplicativeElementWithObjects)",

"CategoryCollections(IsMultiplicativeElementWithObjectsAndOnes)",

"CategoryCollections(IsMultiplicativeElementWithObjectsAndInverses)\

", "CategoryCollections(IsGroupoidElement)", "IsMagmaWithObjects",

"IsMagmaWithObjectsAndOnes", "IsMagmaWithObjectsAndInverses",

"IsGroupoid" ]

gap> FamilyObj( Gd8 ); ## these numbers vary from one run to another

NewFamily( "GroupoidFamily", [ 2722 ], [ 53, 54, 79, 80, 81, 82, 92, 93, 116,

117, 119, 120, 123, 205, 501, 2690, 2703, 2707, 2711, 2715, 2718, 2720,

2721, 2722 ] )

gap> KnownAttributesOfObject( Gd8 );

[ "Name", "Size", "ObjectList", "GeneratorsOfMagmaWithObjects",

"GeneratorsOfGroupoid" ]

gap> KnownTruePropertiesOfObject( Gd8 );

[ "IsNonTrivial", "IsFinite", "IsDuplicateFree", "IsAssociative",

"IsSinglePieceDomain", "IsDirectProductWithCompleteDigraphDomain" ]

gap> RepresentationsOfObject( Gd8 );

[ "IsComponentObjectRep", "IsAttributeStoringRep", "IsMWOSinglePieceRep" ]

gap> RepresentationsOfObject( U5 );

[ "IsComponentObjectRep", "IsAttributeStoringRep", "IsPiecesRep" ]

Similarly, for arrows, we have:

Example

gap> [ a78, e2 ];

[ [m2 : -7 -> -8], [(1,3) : -8 -> -7] ]

gap> CategoriesOfObject(a78);

[ "IsExtLElement", "IsExtRElement", "IsMultiplicativeElement",

"IsMultiplicativeElementWithObjects" ]

gap> FamilyObj( a78 ); ## again these numbers vary

NewFamily( "MultiplicativeElementWithObjectsFamily", [ 2702 ],

[ 79, 80, 81, 82, 116, 119, 122, 2702 ] )

gap> CategoriesOfObject(e2);

[ "IsExtLElement", "IsExtRElement", "IsMultiplicativeElement",

"IsMultiplicativeElementWithObjects",

"IsMultiplicativeElementWithObjectsAndOnes",

"IsMultiplicativeElementWithObjectsAndInverses",

"IsGroupoidElement" ]

gap> FamilyObj( e2 );

NewFamily( "GroupoidElementFamily", [ 2714 ],

[ 79, 80, 81, 82, 116, 119, 122, 2702, 2706, 2710, 2714 ] )

7.2 Many object homomorphisms

Homomorphisms of structures with many objects have a similar heirarchy. A few examples:

� DeclareCategory( "IsGeneralMappingWithObjects", IsGeneralMapping );

� DeclareSynonymAttr( "IsMagmaWithObjectsGeneralMapping",

IsGeneralMappingWithObjects and RespectsMultiplication );

� DeclareSynonymAttr( "IsMagmaWithObjectsHomomorphism",

IsMagmaWithObjectsGeneralMapping and IsMapping );

� DeclareCategory("IsGroupoidHomomorphism",IsMagmaWithObjectsHomomorphism);

Two forms of representation are used: for mappings to a single piece; and for unions of such

mappings:

� DeclareRepresentation( "IsMappingToSinglePieceRep",

IsMagmaWithObjectsHomomorphism and IsAttributeStoringRep and

IsGeneralMapping, [ "Source", "Range", "SinglePieceMappingData" ] );

� DeclareRepresentation( "IsMappingWithObjectsRep",

IsMagmaWithObjectsHomomorphism and IsAttributeStoringRep and

IsGeneralMapping, [ "Source", "Range", "PiecesOfMapping" ] );

In previous chapters, hom1 was an endofunction on M78; homd8 was a homomorphism

from Gd8 to Gs3; and aut3 was an automorphism of Ga4. All homomorphisms have family

GeneralMappingWithObjectsFamily. Perhaps it would be better to have separate families for each

structure?
Example

gap> FamilyObj(hom1);

NewFamily( "GeneralMappingWithObjectsFamily", [ 2726 ],

[ 79, 80, 81, 82, 116, 119, 122, 126, 130, 149, 412, 2726 ] )

gap> KnownAttributesOfObject( hom1 );

[ "Range", "Source", "SinglePieceMappingData" ]

gap> KnownTruePropertiesOfObject( hom1 );

[ "CanEasilyCompareElements", "CanEasilySortElements", "IsTotal",

"IsSingleValued", "RespectsMultiplication", "IsGeneralMappingToSinglePiece",

"IsGeneralMappingFromSinglePiece", "IsInjectiveOnObjects",

"IsSurjectiveOnObjects" ]

gap> CategoriesOfObject( homd8 );

[ "IsExtLElement", "IsExtRElement", "IsMultiplicativeElement",

"IsMultiplicativeElementWithOne", "IsMultiplicativeElementWithInverse",

"IsAssociativeElement", "IsGeneralMapping", "IsGeneralMappingWithObjects",

"IsGroupoidHomomorphism" ]

gap> KnownAttributesOfObject( homd8 );

[ "Range", "Source", "SinglePieceMappingData", "ImagesOfObjects", "ImageElementsOfRays",

"ObjectTransformationOfGroupoidHomomorphism", "RootGroupHomomorphism" ]

gap> KnownAttributesOfObject( aut3 );

[ "Order", "Range", "Source", "SinglePieceMappingData", "ImagesOfObjects",

"ImageElementsOfRays", "ObjectTransformationOfGroupoidHomomorphism",

"RootGroupHomomorphism" ]

Development History

8.1 Versions of the Package

The �rst version,GraphGpd 1.001, formed part of EmmaMoore's thesis [Moo01] in December 2000,

but was not made generally available.

Version 1.002 of GraphGpd was prepared to run under GAP 4.4 in January 2004; was submitted

to the GAP council to be considered as an accepted package; but suggestions from the referee were

not followed up.

In April 2006 the manual was converted to GAPDoc format. Variables Star, Costar and

CoveringGroupwere found to con�ict with usage in other packages, and were renamed VertexStar,

VertexCostar and CoveringGroupOfGroupoid respectively. Similarly, the Vertices and Arcs of

an FpWeightedDigraph were changed from attributes to record components.

In the spring of 2006 the package was extensively rewritten and renamed gpd. Version 1.01

was submitted as a deposited package in June 2006. Version 1.03, of October 2007, �xed some �le

protections, and introduced the test �le gpd_manual.tst.

Version 1.05, of November 2008, was released when the website at Bangor changed.

Since then, the package has been rewritten again, introducing magmas with objects and their

mappings. Functions to implement constructions contained in [AW10] have been added, but this is

ongoing work.

Versions 1.09 to 1.15 were prepared for the anticipated release of GAP 4.5 in June 2012.

gpd became an accepted GAP package in May 2015.

In April 2017 the package was renamed again, as groupoids.

In August 2017 the implementation of groupoid homomorphisms was completely revised with the

emphasis now on a mapping from a set of generating arrows to their images.

In September 2017 various functions were revised so that, at last, the operation

DiscreteNormalPreXModWithObjects in XMod works again. This constructs a crossed module

of groupoids with a connected range and a homogeneous, discrete source.

The latest version is 1.54 of 29th November 2017, for GAP 4.9.

In recent versions there have been a number of changes of function name, such as IsDigraph

becoming GroupoidIsDigraph. This is in order to avoid con�icts with the Digraphs package. The

intention is that these functions can revert to the original names in due course.

8.2 What needs to be done next?

� more work on automorphism groups of groupoids;

� normal subgroupoids and quotient groupoids;

� more methods for morphisms of groupoids, particularly when the range is not connected;

� ImageElm and ImagesSource for the cases of groupoid morphisms not yet covered;

� Enumerator for IsHomsetCosetsRep;

� free groupoid on a graph;

� methods for FreeProductWithAmalgamation and HnnEntension for pc-groups;

� convert GraphOfGroupsRewritingSystem to the category IsRewritingSystem;

� in XMod, continue to work on crossed modules over groupoids.

